

EISDI1 T=00003 IS ON CR00602 USING 00337 BLKS R=0017

```
0001  ASMB,A,B,N,L
0002      IFZ                RELOCATABLE
0003      NAM FDIAG
0004      XIF                END RELOCATABLE
0005      IFN                ABSOLUTE
0006      ORG 0
0007      XIF                END ABSOLUTE
0008      HED FLOPPY DIAGNOSTIC  9/23/76
0009  *
0010  *      AED 6200 FLOPPY DISK DRIVE DIAGNOSTIC
0011  *
0012  *      *PROGRAMMER*
0013  *      DAVE LENNERT                JULY 21, 1976
0014  *      REVISION                SEPTEMBER 23, 1976
0015  *
0016  *
0017  *  THIS DIAGNOSTIC ASSUMES DMA AVAILABLE IN SELECT
0018  *  CODES 2 & 6.
0019  *
0020  *  ALL TESTING (EXCEPT FOR "INITIALIZATION OF USER
0021  *  FLOPPY") MUST BE DONE WITH GOOD MEDIA TO YIELD
0022  *  VALID RESULTS.
0023  *
0024  *  MEDIA CAN BE VERIFIED TO BE GOOD BY INITIALIZING
0025  *  IT USING THE "INITIALIZATION OF USER FLOPPY"
0026  *  SECTION OF THIS DIAGNOSTIC.
0027  *
0028  *
0029  *  THIS DIAGNOSTIC CAN BE LOADED AND RUN AS AN
0030  *  ABSOLUTE PROGRAM, OR UNDER RTE-II [SOFTWARE
0031  *  HANDLING OF MEMORY MAPPING HARDWARE UNDER
0032  *  RTE-III MAKES THE INTERRUPT TEST SECTION OF
0033  *  THIS DIAGNOSTIC INOPERABLE].
0034  *
0035  *  TO ASSEMBLE THIS DIAGNOSTIC AS AN ABSOLUTE
0036  *  PROGRAM, USE THE FOLLOWING FORMAT FOR
0037  *  THE ASMB CARD:
0038  *      ASMB,A,B,N,L
0039  *
0040  *  TO ASSEMBLE THIS DIAGNOSTIC AS A RELOCATABLE
0041  *  PROGRAM (TO LOAD AND START UNDER RTE-II),
0042  *  USE THE FOLLOWING FORMAT FOR THE ASMB CARD:
0043  *      ASMB,R,Z,L
0044  *
0045  *  NOTE:  SW-REG MEANS SWITCH REGISTER
0046  *          A-REG MEANS THE A REGISTER
0047  *          B-REG MEANS THE B REGISTER
0048  *          P-REG MEANS THE P REGISTER
0049  *          T-REG MEANS THE T REGISTER
0050  *      SKP
0051  * *****
0052  *  START UP PROCEDURE FOR RUNNING AS STAND ALONE *
0053  *  AS AN ABSOLUTE PROGRAM: *
0054  * *****
0055  *
0056  *
0057  *  1) ASSEMBLE & LOAD PROGRAM.  (PROGRAM ORIGINATES AT 100B.)
0058  *  2) SET SW-REG BITS 5-0 TO DATA CHANNEL SELECT
0059  *      CODE (THE HIGHEST PRIORITY AND LOWEST
0060  *      NUMERICALLY OF THE TWO SELECT CODES).
0061  *  CLEAR BIT 15 OF THE SW-REG IF FLOPPY DRIVE IS
0062  *  SINGLE DENSITY.
```

```

0063 *   SET BIT 15 OF THE SW-REG IF FLOPPY DRIVE IS
0064 *   DOUBLE DENSITY.
0065 *   SET THE P-REG TO 100 OCTAL.
0066 *   LOAD A GOOD FLOPPY (PREVIOUSLY VALIDATED)
0067 *   IN THE DESIRED UNIT OF THE FLOPPY
0068 *   CONTROLLER.
0069 * 3) PUSH RUN.
0070 * 4) THE COMPUTER SHOULD HALT.
0071 *   THE HALT CODE IS DISPLAYED IN BITS 5-0 OF
0072 *   THE T-REG.
0073 *   LOOK UP THE HALT CODE IN THE HALT CODE
0074 *   TABLE BELOW AND FOLLOW THE INSTRUCTIONS
0075 *   GIVEN THERE.
0076 *   IF AN ERROR HALT IS ENCOUNTERED, DO NOT
0077 *   CONTINUE WITH THE DIAGNOSTIC UNLESS
0078 *   INSTRUCTED TO DO SO.
0079 *
0080 *
0081 *
0082 *****
0083 * START UP PROCEDURE FOR RUNNING UNDER RTE-II: *
0084 *****
0085 *
0086 *
0087 * 1) ASSEMBLE & LOAD PROGRAM UNDER RTE-II.
0088 * 2) SET SW-REG BITS 5-0 TO DATA CHANNEL SELECT
0089 *   CODE (THE HIGHEST PRIORITY AND LOWEST
0090 *   NUMERICALLY OF THE TWO SELECT CODES).
0091 *   CLEAR BIT 15 OF SW-REG IF FLOPPY DRIVE IS
0092 *   SINGLE DENSITY.
0093 *   SET BIT 15 OF SW-REG IF FLOPPY DRIVE IS
0094 *   DOUBLE DENSITY.
0095 *   LOAD A GOOD FLOPPY (PREVIOUSLY VALIDATED)
0096 *   IN THE DESIRED UNIT OF THE FLOPPY
0097 *   CONTROLLER.
0098 * 3) RUN THE PROGRAM. [RU,FDIAG]
0099 * 4) THE COMPUTER SHOULD HALT.
0100 *   THE HALT CODE IS DISPLAYED IN BITS 5-0 OF
0101 *   THE T-REG.
0102 *   LOOK UP THE HALT CODE IN THE HALT CODE
0103 *   TABLE BELOW AND FOLLOW THE INSTRUCTIONS
0104 *   GIVEN THERE.
0105 *   IF AN ERROR HALT IS ENCOUNTERED, DO NOT
0106 *   CONTINUE WITH THE DIAGNOSTIC UNLESS
0107 *   INSTRUCTED TO DO SO.
0108 * 5) WHEN FINISHED WITH DIAGNOSTIC, DO A COLD
0109 *   BOOT TO GET BACK TO RTE. (THE DIAGNOSTIC
0110 *   OVERWRITES PART OF RTE.)
0111 *
0112 *
0113 *   SKP
0114 *****
0115 * HALT CODES (OCTAL) *
0116 *****
0117 *
0118 * 00 I/O INSTRUCTION CONFIGURATION COMPLETE.
0119 *   SET BIT 15 OF SW-REG TO LOOP ON CONTROL &
0120 *   FLAG BITS AND INTERRUPT TEST.
0121 *   SET BIT 0 OF SW-REG TO GO TO
0122 *   "INITIALIZATION OF USER FLOPPY"
0123 *   SECTION OF DIAGNOSTIC.
0124 *   CLEAR BOTH BITS 15 & 0 OF SW-REG FOR
0125 *   NORMAL DIAGNOSTIC.
0126 *   PUSH RUN.

```

```

0127 *
0128 * *****
0129 * * CONTROL & FLAG BITS AND *
0130 * * INTERRUPT TEST *
0131 * *****
0132 *
0133 * 01 ERROR:
0134 * DATA CHANNEL FLAG WILL NOT CLEAR.
0135 *
0136 * 02 ERROR:
0137 * CMND CHANNEL FLAG WILL NOT CLEAR.
0138 *
0139 * 03 ERROR:
0140 * DATA CHANNEL FLAG WILL NOT SET.
0141 *
0142 * 04 ERROR:
0143 * DATA CONTROL WILL NOT CLEAR AND/OR DATA
0144 * INTERRUPT REQUEST LINE STUCK ON.
0145 *
0146 * 05 ERROR:
0147 * DATA CONTROL WILL NOT SET AND/OR DATA
0148 * CHANNEL WILL NOT REQUEST INTERRUPT.
0149 *
0150 * 06 ERROR:
0151 * CMND CHANNEL FLAG WILL NOT SET.
0152 *
0153 * 07 ERROR:
0154 * CMND CONTROL WILL NOT CLEAR AND OR CMND
0155 * INTERRUPT REQUEST LINE STUCK ON.
0156 *
0157 * 10 ERROR:
0158 * CMND CONTROL WILL NOT SET AND/OR CMND
0159 * CHANNEL WILL NOT REQUEST INTERRUPT.
0160 *
0161 * 11 SUCCESSFUL COMPLETION OF CONTROL & FLAG
0162 * BITS AND INTERRUPT TESTS.
0163 * CLEAR BIT 15 OF SW-REG TO CONTINUE WITH
0164 * DIAGNOSTIC.
0165 * SET BIT 15 OF SW-REG TO LOOP ON CONTROL
0166 * & FLAG BITS AND INTERRUPT TESTS.
0167 * PUSH RUN.
0168 *
0169 * SKP
0170 * *****
0171 * * DRIVE DENSITY TEST *
0172 * *****
0173 *
0174 * 12 ERROR:
0175 * DENSITY BIT FROM FLOPPY STATUS WORD (BIT 5:
0176 * 0-SINGLE, 1-DOUBLE) DOES NOT AGREE WITH
0177 * USER SUPPLIED INFORMATION.
0178 * FLOPPY STATUS WORD IS IN A-REG.
0179 * PUSH RUN TO CONTINUE WITH DIAGNOSTIC AND
0180 * DENSITY WILL BE ASSUMED TO BE WHATEVER
0181 * THE USER SPECIFIED AT DIAGNOSTIC START UP.
0182 *
0183 * 13 ENTER FLOPPY UNIT SELECT NUMBER (0-3) IN
0184 * BITS 1-0 OF SW-REG.
0185 * CLEAR BIT 15 OF SW-REG TO CONTINUE WITH
0186 * NORMAL DIAGNOSTIC.
0187 * SET BIT 15 OF SW-REG TO GO TO
0188 * "INITIALIZATION OF USER FLOPPY"
0189 * SECTION OF DIAGNOSTIC.
0190 * ENABLE INIT SWITCH ON FRONT PANEL.

```

```

0191 *      PUSH RUN TO CONTINUE WITH DIAGNOSTIC.
0192 *
0193 *      *****
0194 *      * ADDR FIELD TEST *
0195 *      *****
0196 *
0197 * 14  ERROR:
0198 *      INITIALIZE ERROR BIT IN STATUS WORD
0199 *      SET WHEN ATTEMPTED TO ENTER INITIALIZE
0200 *      MODE.
0201 *      IF INIT SWITCH ON FRONT PANEL IS ENABLED,
0202 *      THIS IS AN ERROR.
0203 *      IF INIT SWITCH IS DISABLED, THEN ENABLE IT
0204 *      AND SET BIT 15 OF THE SW-REG. (THIS
0205 *      WILL CAUSE THE ADDR FIELD TEST
0206 *      TO LOOP. CLEAR BIT 15 OF THE SW-REG
0207 *      AFTER A SUFFICIENT AMOUNT OF TIME TO
0208 *      ALLOW AT LEAST ONE LOOP.)
0209 *      PUSH RUN.
0210 *
0211 * 15  ERROR:
0212 *      INITIALIZE MODE BIT IN STATUS WORD
0213 *      FAILED TO SET AFTER AN ATTEMPT TO
0214 *      ENTER INITIALIZE MODE.
0215 *
0216 * 16  ERROR:
0217 *      INITIALIZE FLAG IN STATUS WORD FAILED
0218 *      TO CLEAR WHILE COUNTING ZEROS DURING
0219 *      INITIALIZATION.
0220 *
0221 * 17  ERROR:
0222 *      INITIALIZE FLAG IN STATUS WORD FAILED
0223 *      TO SET WHILE COUNTING ZEROS DURING
0224 *      INITIALIZATION.
0225 *
0226 * 20  ERROR:
0227 *      TRACK + LOSS WORD READ FROM ADDR FIELD
0228 *      DOES NOT AGREE WITH WHAT WAS WRITTEN.
0229 *      THE A-REG CONTAINS THE TRACK + LOSS
0230 *      WORD READ FROM THE ADDR FIELD.
0231 *      THE B-REG CONTAINS THE TRACK + LOSS
0232 *      WORD WRITTEN.
0233 *      PUSH RUN TO TEST SECTOR + HOSS WORD.
0234 *
0235 * 21  ERROR:
0236 *      SECTOR + HOSS WORD READ FROM ADDR FIELD
0237 *      DOES NOT AGREE WITH WHAT WAS WRITTEN.
0238 *      THE A-REG CONTAINS THE SECTOR + HOSS
0239 *      WORD READ FROM THE ADDR FIELD.
0240 *      THE B-REG CONTAINS THE SECTOR + HOSS
0241 *      WORD WRITTEN.
0242 *      PUSH RUN IF WANT TO LOOP ON ADDR FIELD
0243 *      TEST.
0244 *
0245 * 22  COMPLETION OF ADDR FIELD TEST.
0246 *      IF NO ERROR HALTS WERE ENCOUNTERED,
0247 *      INDICATES SUCCESSFUL COMPLETION.
0248 *      CLEAR BIT 15 OF THE SW-REG TO CONTINUE
0249 *      WITH DIAGNOSTIC.
0250 *      SET BIT 15 OF THE SW-REG TO LOOP ON
0251 *      ADDR FIELD TEST.
0252 *      ENABLE INIT SWITCH ON FRONT PANEL.
0253 *      PUSH RUN.
0254 *

```

```

0255 * *****
0256 * * SECTOR FIELD TEST *
0257 * *****
0258 *
0259 * 23 ERROR:
0260 * DATA READ FROM SECTOR DOES NOT AGREE
0261 * WITH THAT WRITTEN.
0262 * THE A-REG CONTAINS THE NUMBER OF WORDS
0263 * WRITTEN.
0264 * THE B-REG CONTAINS THE NUMBER OF WORDS
0265 * READ WHICH WERE INCORRECT.
0266 * CLEAR BIT 15 OF THE SW-REG TO CONTINUE
0267 * WITH DIAGNOSTIC.
0268 * SET BIT 15 OF THE SW-REG TO LOOP ON
0269 * SECTOR FIELD TEST.
0270 * PUSH RUN.
0271 *
0272 * 24 SUCCESSFUL COMPLETION OF SECTOR FIELD
0273 * TEST.
0274 * CLEAR BIT 15 OF THE SW-REG TO CONTINUE
0275 * WITH DIAGNOSTIC.
0276 * SET BIT 15 OF THE SW-REG TO LOOP ON
0277 * SECTOR FIELD TEST.
0278 * ENABLE INIT SWITCH ON FRONT PANEL.
0279 * PUSH RUN.
0280 *
0281 * *****
0282 * * TEST INITIALIZATION OF FLOPPY *
0283 * *****
0284 *
0285 * 25 ERROR:
0286 * ERROR OCCURRED WHILE READING BACK A
0287 * SECTOR WHICH WAS INITIALIZED.
0288 * CHECK THE FLOPPY STATUS WORD IN THE
0289 * B-REG TO TELL WHAT THE SPECIFIC
0290 * ERROR WAS.
0291 * THE A-REG CONTAINS THE TRACK ADDR
0292 * BEING READ IN BITS 15-8, AND THE
0293 * SECTOR ADDR IN BITS 7-0.
0294 * THE B-REG CONTAINS THE FLOPPY STATUS WORD.
0295 *
0296 * 26 ERROR:
0297 * DATA READ IN FROM AN INITIALIZED SECTOR
0298 * DOES NOT AGREE WITH WHAT WAS WRITTEN.
0299 * DATA WORD SHOULD HAVE BEEN OCTAL 066666.
0300 * THE A-REG CONTAINS THE TRACK ADDR BEING
0301 * CHECKED IN BITS 15-8, AND THE SECTOR
0302 * ADDR IN BITS 7-0.
0303 * THE B-REG CONTAINS THE DATA WORD WHICH
0304 * DOES NOT MATCH.
0305 *
0306 * 27 ERROR:
0307 * INITIALIZE MODE CANNOT BE ENTERED.
0308 * IF INIT SWITCH ON FRONT PANEL IS ENABLED,
0309 * THEN THIS IS AN ERROR.
0310 * IF INIT SWITCH IS DISABLED, ENABLE IT AND
0311 * PUSH RUN TO RETRY.
0312 *
0313 * 30 SUCCESSFUL COMPLETION OF TEST INITIALIZATION
0314 * OF FLOPPY.
0315 * CLEAR BIT 15 OF SW-REG TO CONTINUE
0316 * WITH NORMAL DIAGNOSTIC.
0317 * SET BIT 15 OF SW-REG TO LOOP ON
0318 * CRC DETECTION TEST.

```

```

0319 *      ENABLE INIT SWITCH ON FRONT PANEL.
0320 *      PUSH RUN TO CONTINUE WITH DIAGNOSTIC.
0321 *
0322 *      !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
0323 *      ! NOTE: THE "CRC DETECTION CHECK" SECTION !
0324 *      ! OF THE DIAGNOSTIC IS JUMPED OVER DUE !
0325 *      ! HARDWARE PROBLEMS BEYOND OUR (AND !
0326 *      ! PROBABLY AED'S) CONTROL. !
0327 *      ! AFTER THE END OF THE ABOVE DIAGNOSTIC !
0328 *      ! SECTION, CONTROL GOES IMMEDIATELY TO !
0329 *      ! THE SECTION FOLLOWING THE "CRC !
0330 *      ! DETECTION CHECK" SECTION. !
0331 *      !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
0332 *
0333 *      *****
0334 *      * CRC DETECTION CHECK *
0335 *      *****
0336 *
0337 * 31  ERROR:
0338 *      INITIALIZE MODE CANNOT BE ENTERED.
0339 *      IF INIT SWITCH ON FRONT PANEL IS ENABLED,
0340 *      THEN THIS IS AN ERROR.
0341 *      IF INIT SWITCH IS DISABLED, ENABLE IT AND
0342 *      PUSH RUN TO RETRY.
0343 *
0344 * 32  ERROR:
0345 *      CRC ERROR DETECTION FAILED.
0346 *      DATA CHECK ERROR BIT FAILED TO SET
0347 *      ON A FORCED DATA ERROR.
0348 *      THE B-REG CONTAINS THE FLOPPY STATUS WORD.
0349 *      CLEAR BIT 15 OF SW-REG TO CONTINUE
0350 *      WITH DIAGNOSTIC.
0351 *      SET BIT 15 OF SW-REG TO LOOP ON
0352 *      CRC DETECTION CHECK.
0353 *      PUSH RUN.
0354 *
0355 * 33  SUCCESSFUL COMPLETION OF CRC DETECTION
0356 *      CHECK.
0357 *      THE B-REG CONTAINS THE FLOPPY STATUS WORD.
0358 *      CLEAR BIT 15 OF SW-REG TO CONTINUE
0359 *      WITH DIAGNOSTIC.
0360 *      SET BIT 15 OF SW-REG TO LOOP ON
0361 *      CRC DETECTION CHECK.
0362 *      ENABLE INIT SWITCH ON FRONT PANEL.
0363 *      PUSH RUN.
0364 *
0365 *      *****
0366 *      * READ/WRITE HARDWARE TEST *
0367 *      *****
0368 *
0369 * 34  ERROR:
0370 *      INITIALIZE MODE CANNOT BE ENTERED.
0371 *      IF INIT SWITCH ON FRONT PANEL IS UP,
0372 *      THEN THIS IS AN ERROR.
0373 *      IF INIT SWITCH IS DISABLED, ENABLE IT AND
0374 *      PUSH RUN TO RETRY.
0375 *
0376 * 35  ERROR:
0377 *      ADDRESS CHECK OCCURRED ON FLOPPY READ.
0378 *      PROBABLE HARDWARE ERROR.
0379 *      THE A-REG CONTAINS THE TRACK ADDR BEING
0380 *      READ IN BITS 15-8, AND THE SECTOR ADDR
0381 *      IN BITS 7-0.
0382 *      THE B-REG CONTAINS THE FLOPPY STATUS WORD.

```

```

0383 *      CLEAR BIT 15 OF THE SW-REG TO EXIT
0384 *      READ/WRITE HARDWARE TEST AND CONTINUE
0385 *      WITH REMAINING DIAGNOSTIC.
0386 *      SET BIT 15 OF THE SW-REG TO CONTINUE
0387 *      WITH READ/WRITE HARDWARE TEST.
0388 *      PUSH RUN.
0389 *
0390 * 36  ERROR:
0391 *      DATA READ DOES NOT AGREE WITH DATA WRITTEN.
0392 *      THE A-REG CONTAINS THE TRACK ADDR BEING
0393 *      CHECKED IN BITS 15-8, AND THE SECTOR
0394 *      ADDR IN BITS 7-0.
0395 *      THE B-REG CONTAINS THE DATA WORD WHICH
0396 *      DOES NOT MATCH.  THE DATA WHICH WAS
0397 *      WRITTEN WAS AN ALTERNATING BIT PATTERN,
0398 *      EITHER OCTAL 052525 OR 125252.
0399 *      PUSH RUN TO CONTINUE WITH READ/WRITE
0400 *      HARDWARE TEST.
0401 *
0402 * 37  ERROR:
0403 *      ADDRESS CHECK OCCURRED ON FLOPPY WRITE.
0404 *      PROBABLE HARDWARE ERROR.
0405 *      THE A-REG CONTAINS THE TRACK ADDR BEING
0406 *      WRITTEN IN BITS 15-8, AND THE SECTOR
0407 *      ADDR IN BITS 7-0.
0408 *      THE B-REG CONTAINS THE FLOPPY STATUS WORD.
0409 *      CLEAR BIT 15 OF THE SW-REG TO EXIT
0410 *      READ/WRITE HARDWARE TEST AND CONTINUE
0411 *      WITH REMAINING DIAGNOSTIC.
0412 *      SET BIT 15 OF THE SW-REG TO CONTINUE
0413 *      WITH READ/WRITE HARDWARE TEST.
0414 *      PUSH RUN.
0415 *
0416 * 40  COMPLETION OF READ/WRITE HARDWARE TEST.
0417 *      IF NO ERROR HALTS WERE ENCOUNTERED
0418 *      BEFORE THIS HALT THEN THIS INDICATES
0419 *      SUCCESSFUL COMPLETION OF READ/WRITE
0420 *      HARDWARE TEST.
0421 *      PUSH RUN TO CONTINUE WITH DIAGNOSTIC.
0422 *
0423 *
0424 *
0425 * *****
0426 * * INITIALIZATION OF USER FLOPPY *
0427 * *****
0428 *
0429 * 50  BEGINNING OF USER FLOPPY INITIALIZATION.
0430 *      ENTER INITIALIZATION STARTING TRACK ADDR
0431 *      (INCLUSIVE) IN SW-REG.
0432 *      PUSH RUN.
0433 *
0434 * 51  ENTER INITIALIZATION ENDING TRACK ADDR
0435 *      (INCLUSIVE) IN SW-REG.
0436 *      PUSH RUN.
0437 *
0438 * 52  ENTER # SECTORS/TRACK FOR THE ABOVE
0439 *      SPECIFIED TRACK REGION IN SW-REG.
0440 *      PUSH RUN.
0441 *
0442 * 53  ENTER # WORDS/SECTOR FOR THE ABOVE
0443 *      SPECIFIED TRACK REGION IN SW-REG.
0444 *      PUSH RUN.
0445 *
0446 * 54  ENTER INTERLACING SKIP COUNT FOR THE

```

```

0447 *      ABOVE SPECIFIED TRACK REGION IN SW-REG.
0448 *      NOTE: THE # SECTORS/TRACK MUST BE AN
0449 *      INTEGER MULTIPLE OF THE SKIP COUNT + 1.
0450 *      PUSH RUN.
0451 *
0452 * 55  ENTER DATA WORD TO BE WRITTEN INTO
0453 *      SECTORS FOR THE ABOVE SPECIFIED TRACK
0454 *      REGION IN SW-REG.
0455 *      MAKE SURE THE INIT SWITCH ON THE FRONT
0456 *      PANEL IS ENABLED.
0457 *      PUSH RUN.
0458 *
0459 * 56  ERROR:
0460 *      INVALID TRACK ADDR:
0461 *      (STARTING TRACK ADDR) < 0 OR > 76
0462 *      (ENDING TRACK ADDR)  < 0 OR > 76
0463 *      (STARTING TRACK ADDR) > (ENDING TRACK
0464 *      ADDR)
0465 *      PUSH RUN TO RE-ENTER PARAMETERS.
0466 *
0467 * 57  ERROR:
0468 *      # SECTORS/TRACK [HALT 52] DIVIDED BY
0469 *      INTERLACING SKIP COUNT [HALT 54] IS
0470 *      NOT AN INTEGER.
0471 *      PUSH RUN TO RE-ENTER PARAMETERS.
0472 *
0473 * 60  ERROR:
0474 *      COMPUTED TRACK SIZE IS TOO BIG.
0475 *      # SECTORS/TRACK AND/OR # WORDS/SECTOR
0476 *      MUST BE ADJUSTED ACCORDINGLY.
0477 *      PUSH RUN TO RE-ENTER PARAMETERS.
0478 *
0479 * 61  ERROR:
0480 *      INITIALIZATION MODE CANNOT BE ENTERED.
0481 *      CHECK INIT SWITCH ON FRONT PANEL, IT
0482 *      SHOULD BE ENABLED.
0483 *      IF INIT SWITCH IS ALREADY ENABLED THEN A
0484 *      HARDWARE FAILURE IS INDICATED,
0485 *      INITIALIZE MODE BIT (BIT 4) OF THE
0486 *      FLOPPY STATUS WORD WILL NOT SET.
0487 *      THE A-REG CONTAINS THE CURRENT TRACK ADDR
0488 *      BEING INITIALIZED IN BITS 15-8, AND
0489 *      THE NEXT SECTOR ADDR IN BITS 7-0.
0490 *      SET BIT 15 OF THE SW-REG TO RESTART THE
0491 *      INITIALIZATION PROCESS USING THE SAME
0492 *      PARAMETERS.
0493 *      CLEAR BIT 15 OF THE SW-REG TO RE-ENTER
0494 *      NEW INITIALIZATION PARAMETERS.
0495 *      PUSH RUN.
0496 *
0497 * 62  ERROR:
0498 *      ERROR OCCURRED ON A READ TO CHECK
0499 *      INITIALIZED AREA OF FLOPPY. PROBABLE
0500 *      CAUSE: BAD MEDIA.
0501 *      CHECK THE FLOPPY STATUS WORD IN THE
0502 *      B-REG TO TELL WHAT THE SPECIFIC
0503 *      ERROR WAS.
0504 *      THE A-REG CONTAINS THE TRACK ADDR BEING
0505 *      READ IN BITS 15-8, AND THE SECTOR
0506 *      ADDR IN BITS 7-0.
0507 *      THE B-REG CONTAINS THE FLOPPY STATUS WORD.
0508 *      CLEAR BIT 15 OF THE SW-REG TO CONTINUE
0509 *      CHECKING AT THE NEXT SECTOR.
0510 *      SET BIT 15 OF THE SW-REG TO ABORT FURTHER

```

```

0511 *      MEDIA CHECKING.
0512 *      PUSH RUN.
0513 *
0514 * 63  ERROR:
0515 *      DATA READ BACK FROM INITIALIZED SECTOR
0516 *      DOES NOT AGREE WITH WHAT WAS WRITTEN
0517 *      (SPECIFIED AT HALT 55).
0518 *      THE A-REG CONTAINS THE TRACK ADDR BEING
0519 *      CHECKED IN BITS 18-8, AND THE SECTOR
0520 *      ADDR IN BITS 7-0.
0521 *      THE B-REG CONTAINS THE DATA WORD WHICH
0522 *      DOES NOT MATCH.
0523 *      CLEAR BIT 15 OF THE SW-REG TO CONTINUE
0524 *      CHECKING AT THE NEXT SECTOR.
0525 *      SET BIT 15 OF THE SW-REG TO ABORT FURTHER
0526 *      MEDIA CHECKING.
0527 *      PUSH RUN.
0528 *
0529 * 64  INITIALIZATION OF THE SPECIFIED TRACK
0530 *      REGION COMPLETED. IF NO ERROR HALTS
0531 *      WERE ENCOUNTERED THEN INITIALIZATION
0532 *      WAS SUCCESSFUL AND THE MEDIA IS GOOD.
0533 *      SET BIT 15 OF THE SW-REG TO RESTART THE
0534 *      INITIALIZATION PROCESS USING THE SAME
0535 *      PARAMETERS.
0536 *      CLEAR BIT 15 OF THE SW-REG TO RE-ENTER
0537 *      NEW INITIALIZATION PARAMETERS.
0538 *      PUSH RUN.
0539 *
0540 *
0541 *      *****
0542 *      * GLOBAL ERRORS *
0543 *      *****
0544 *
0545 * 73  ERROR:
0546 *      CMND FLAG TIMED OUT:
0547 *      CMND FLAG DID NOT SET WHEN EXPECTED.
0548 *      THE A-REG CONTAINS THE CURRENT TRACK ADDR
0549 *      IN BITS 15-8, AND THE NEXT SECTOR
0550 *      ADDR IN BITS 7-0.
0551 *      THE B-REG CONTAINS THE FLOPPY STATUS WORD.
0552 *
0553 * 74  ERROR:
0554 *      CONTINUITY CHECKS IN THE DIAGNOSTIC HAVE
0555 *      FAILED. WHAT CAN I SAY?
0556 *      PROBABLY CAUSE: USER MONKEYING WITH CORE.
0557 *
0558 *      HED FLOPPY DIAGNOSTIC - SETUP
0559 *      A      EQU 0
0560 *      B      EQU 1
0561 *      SWREG EQU 1          SWITCH REGISTER SELECT CODE
0562 *      DATA EQU 0
0563 *      CMND  EQU 0
0564 *      SPC    1
0565 *      *****
0566 *
0567 *      START UP PROCEDURE FOR RUNNING UNDER RTE-II
0568 *
0569 *      *****
0570 *      SPC 1
0571 *      IFZ          RELOCATABLE
0572 *      EXT $LIBR
0573 *      RTE JSB $LIBR  TURN OFF INTERRUPT SYSTEM
0574 *      NOP

```

```

0575          SPC 1
0576          CLF 00          REALLY TURN OFF INTERRUPT SYSTEM
0577          SPC 1
0578          *****
0579          *
0580          * FORCE LOAD SOME BASE PAGE LINKAGES
0581          *
0582          *****
0583          SPC 1
0584          LDA ATST1
0585          STA BTST1
0586          LDA ATST2
0587          STA BTST2
0588          LDA ATST3
0589          STA BTST3
0590          LDA ATST4
0591          STA BTST4
0592          SPC 1
0593          XIF          END RELOCATABLE
0594          SPC 1
0595          *****
0596          *
0597          * START IS WHERE THE USER STARTS THIS DIAGNOSTIC
0598          * IF HE IS RUNNING IT ABSOLUTE.
0599          *
0600          *****
0601          SPC 1
0602          START EQU 100B
0603          SPC 1
0604          IFN          ABSOLUTE
0605          ORG START    USER STARTUP LOCATION
0606          XIF          END ABSOLUTE
0607          SPC 1
0608          JMP GO       GO TO PROGRAM BEGINNING
0609          SPC 2
0610          *****
0611          *
0612          * BTST1 IS WHERE LINKS FOR INTERRUPT INSTRUCTIONS
0613          * ARE STORED.
0614          *
0615          *****
0616          SPC 1
0617          IFN          ABSOLUTE
0618          BTST1 EQU *   IF ABS THEN ON BASE PAGE ALREADY
0619          XIF          END ABSOLUTE
0620          SPC 1
0621          IFZ          RELOCATABLE
0622          BTST1 EQU 100B IF REL THEN FORCE TO BASE PAGE
0623          XIF          END RELOCATABLE
0624          SPC 1
0625          BTST2 EQU BTST1+1
0626          BTST3 EQU BTST1+2
0627          BTST4 EQU BTST1+3
0628          SPC 1
0629          *****
0630          *
0631          * HERE ARE LINKS FOR INTERRUPT INSTRUCTIONS.
0632          * THEY ARE LOADED ON THE BASE PAGE STATICALLY
0633          * FOR ABSOLUTE OR DYNAMICALLY FOR RELOCATABLE.
0634          *
0635          *****
0636          SPC 2
0637          ATST1 DEF TST1
0638          ATST2 DEF TST2

```

```

0639 ATST3 DEF TST3
0640 ATST4 DEF TST4
0641 SPC 1
0642 *****
0643 *
0644 * THESE ARE LINKS TO SUBROUTINES.....
0645 *
0646 *****
0647 SPC 1
0648 .READ DEF READ
0649 .WRIT DEF WRITE
0650 .CHCK DEF CHECK
0651 SPC 1
0652 ***
0653 * CONSTANTS
0654 ***
0655 SPC 1
0656 B1K OCT 1000
0657 B1004 OCT 1004
0658 B14K OCT 14000
0659 B177 OCT 177
0660 B.177 OCT 177700
0661 B2K OCT 2000
0662 B20K OCT 20000
0663 B24K OCT 24000
0664 B3K OCT 3000
0665 B30K OCT 30000
0666 B34K OCT 34000
0667 B37K3 OCT 37030
0668 B377 OCT 377
0669 B40 OCT 40
0670 B4K OCT 4000
0671 ALTB OCT 52525 ALTERNATING BITS
0672 HSIX OCT 66666 HARD SIX'S
0673 B77 OCT 77
0674 DM10 DEC -10
0675 DM1K DEC -1000
0676 DM2 DEC -2
0677 DM23 DEC -23
0678 DM5 DEC -5
0679 DM5.2 DEC -5200
0680 DM5.5 DEC -5500
0681 D10 DEC 10
0682 D1K DEC 1000
0683 D1.5 DEC 1500
0684 D2 DEC 2
0685 D22 DEC 22
0686 D3 DEC 3
0687 D4 DEC 4
0688 D5 DEC 5
0689 D76 DEC 76
0690 SKP
0691 HED FLOPPY DIAGNOSTIC - DATA AREA
0692 *****
0693 * DATA AREA *
0694 *****
0695 SPC 1
0696 CPTR DEF CTB.I CMND INST TABLE PTR
0697 DPTR DEF DTB.I DATA INST TABLE PTR
0698 SPC 1
0699 DD DEF *-* NON-ZERO IF DOUBLE DENSITY
0700 UNITH DEF *-* FLOPPY UNIT # IN BITS 15-14
0701 TRACK DEF *-* READ/WRITE TRACK ADDR
0702 SECTR DEF *-* READ/WRITE SECTOR ADDR

```

```

0703 STRK DEF ** STARTING TRACK ADDR
0704 ETRK DEF ** ENDING TRACK ADDR
0705 CTRK DEF ** CURRENT TRACK ADDR
0706 SECL DEF ** # WORDS/SECTOR
0707 NSEC DEF ** # SECTORS/TRACK
0708 SKIP DEF ** SECTOR INTERLACE FACTOR
0709 WORD DEF ** DATA WORD
0710 ABUF DEF BUF ADDR OF BUFFER
0711 BUFL DEF ** BUFFER LENGTH
0712 COUNT DEF ** COUNTER STORAGE
0713 CNT1 DEF ** COUNTER STORAGE (READ/WRITE TEST)
0714 CW2 DEF ** COMMAND WORD 2
0715 CW3 DEF ** COMMAND WORD 3
0716 BYTES DEF ** # BYTES IN SECTOR
0717 SCHOS DEF ** SECTOR + HOSS
0718 TRLOS DEF ** TRACK + LOSS
0719 SPC 1
0720 JT1 JMP BTST1,I INTERRUPT VECTOR
0721 JT2 JMP BTST2,I INSTRUCTIONS
0722 JT3 JMP BTST3,I
0723 JT4 JMP BTST4,I
0724 SPC 1
0725 TEMP1 DEF ** TEMPORARY STORAGE
0726 HED FLOPPY DIAGNOSTIC - CONFIGURE I/O INSTRUCTIONS
0727 *****
0728 *
0729 * SW-REG = SELECT CODE OF DATA CHANNEL IN BITS 6-0 *
0730 * (SELECT CODE OF CMND CHANNEL = SELECT *
0731 * CODE OF DATA CHANNEL + 1) *
0732 *
0733 * SET BIT 15 IF FLOPPY DRIVE IS DOUBLE DENSITY. *
0734 * CLEAR BIT 15 IF FLOPPY DRIVE IS SINGLE DENSITY. *
0735 *
0736 *****
0737 SPC 2
0738 GO EQU *
0739 CLC 00,C TURN OFF INT. & DEVICES
0740 LIA SWREG LOAD SW-REG CONTENTS
0741 SPC 1
0742 CLB B-REG = 0
0743 SSA IF BIT 15 IS SET THEN
0744 LDB B40 LOAD THE DOUBLE DENSITY MASK
0745 STB DD SAVE AS DOUBLE DENSITY FLAG
0746 SPC 1
0747 AND B77 ISOLATE S.C. IN BITS 6-0
0748 SWP SWAP REGISTERS
0749 SPC 1
0750 *****
0751 *
0752 * CONFIGURE I/O INSTRUCTIONS
0753 *
0754 *****
0755 SPC 1
0756 LDA DPTR LOAD DATA INST TABLE PTR
0757 STA TEMP1 SAVE IN TEMP1
0758 SPC 1
0759 DCON EQU *
0760 LDA TEMP1,I LOAD DATA I/O INST
0761 SZA,RSS IF TABLE EOT THEN
0762 JMP OUT1 EXIT
0763 AND B.177 ZERO S.C. FIELD
0764 IOR B INSERT DATA S.C.
0765 STA TEMP1,I REPLACE INST IN MEMORY
0766 ISZ TEMP1 BUMP TABLE PTR

```

```

0767      JMP DCON      LOOP
0768      SPC 1
0769  OUT1  EQU *
0770      SPC 1
0771      LDA CPTR      LOAD CMND INST TABLE PTR
0772      STA TEMP1     SAVE IN TEMP1
0773      INB           BUMP DATA S.C. TO CMND
0774      SPC 1
0775  CCON  EQU *
0776      LDA TEMP1,I   LOAD CMND I/O INST
0777      SZA,RSS       IF TABLE EOT THEN
0778      JMP OUT2      EXIT
0779      AND B.177     ZERO S.C. FIELD
0780      IOR B         INSERT CMND S.C.
0781      STA TEMP1,I   REPLACE INST IN MEMORY
0782      ISZ TEMP1     BUMP TABLE PTR
0783      JMP CCON      LOOP
0784      SPC 1
0785  OUT2  EQU *
0786      SPC 1
0787      HLT 00B       HALT FOR DATA *****
0788      SPC 1
0789      LIA SWREG     LOAD SW-REG
0790      SLA           IF BIT 0 SET THEN
0791      JMP FINIT     GO TO USER INIT.
0792      HED FLOPPY DIAG. - TEST CONTROL & FLAG BITS AND INTERRUPTS
0793  *****
0794  *
0795  *   CLEAR BIT 15 OF SW-REG FOR NORMAL DIAGNOSTIC.
0796  *   SET BIT 15 OF SW-REG TO LOOP ON CONTROL &
0797  *   FLAG BITS AND INTERRUPT TEST.
0798  *
0799  *****
0800      SPC 1
0801  TL1   EQU *
0802      LIB SWREG     LOAD SW-REG
0803  G.D1  CLC DATA,C CLEAR DATA CONTROL & FLAG
0804      SSB           IF BIT 15 OF SW-REG SET THEN
0805      JMP *+3       SKIP TEST
0806  G.D2  SFC DATA   IF DATA FLAG SET THEN
0807      HLT 01B       ERROR HALT *****
0808      SPC 1
0809  G.C1  CLC CMND,C  CLEAR CMND CONTROL & FLAG
0810      SSB           IF BIT 15 OF SW-REG SET THEN
0811      JMP *+3       SKIP TEST
0812  G.C2  SFC CMND    IF CMND FLAG SET THEN
0813      HLT 02B       ERROR HALT *****
0814      SPC 1
0815      CLF 00        DISABLE INTERRUPTS
0816      SPC 1
0817  G.D3  STF DATA   SET DATA FLAG
0818      SSB           IF BIT 15 OF SW-REG SET THEN
0819      JMP *+3       SKIP TEST
0820  G.D4  SFS DATA   IF DATA FLAG NOT SET THEN
0821      HLT 03B       ERROR HALT *****
0822      SPC 1
0823      LDA JT1       LOAD THE DATA INT. INST.
0824  G.D5  STA DATA   STORE IT
0825      CLA           NOP THE CMND INT.
0826  G.C3  STA CMND    LOCATION
0827      SPC 1
0828      STF 00        TURN ON INTERRUPTS
0829      NOP          WAIT
0830      NOP          WAIT

```

```

0831      NOP      WAIT
0832      JMP CON1  GOOD THING: NO INTERRUPT
0833      SPC 1
0834      *****
0835      *
0836      * ENTER IF DATA INT. WITH FLAG SET AND
0837      * CONTROL SUPPOSEDLY CLEAR.
0838      * -> CONTROL DOES NOT CLEAR OR INT. REQUEST
0839      * LINE STUCK ON!
0840      *
0841      *****
0842      SPC 1
0843      TST1 EQU *
0844      SSB,RSS      IF BIT 15 OF SW-REG CLEAR THEN
0845      HLT 04B      ERROR HALT *****
0846      SPC 1
0847      CON1 EQU *
0848      LDA JT2      RESET THE DATA INT.
0849      G.D6 STA DATA LOCATION
0850      SPC 1
0851      G.D7 STC DATA FORCE DATA INTERRUPT
0852      NOP      WAIT
0853      NOP      WAIT
0854      NOP      WAIT
0855      SSB,RSS      IF BIT 15 OF SW-REG CLEAR THEN
0856      HLT 05B      ERROR HALT *****
0857      SPC 1
0858      *****
0859      *
0860      * ENTER ON EXPECTED INTERRUPT FROM DATA CHAN
0861      *
0862      *****
0863      SPC 1
0864      TST2 EQU *
0865      SPC 1
0866      CLF 00      DISABLE INTERRUPTS
0867      G.D8 CLC DATA,C CLEAR DATA CONTROL & FLAG
0868      SPC 1
0869      G.C4 STF CMND SET CMND FLAG
0870      SSB      IF BIT 15 OF SW-REG SET THEN
0871      JMP ++3      SKIP TEST
0872      G.C10 SFS CMND IF CMND FLAG NOT SET THEN
0873      HLT 06B      ERROR HALT *****
0874      SPC 1
0875      LDA JT3      SETUP CMND INT.
0876      G.C5 STA CMND LOCATION
0877      CLA      NOP THE DATA INT.
0878      G.D9 STA DATA LOCATION
0879      SPC 1
0880      STF 00      ENABLE INTERRUPTS
0881      NOP      WAIT
0882      NOP      WAIT
0883      NOP      WAIT
0884      JMP CON2      OKY-DOKY: NO INTERRUPT
0885      SPC 1
0886      *****
0887      *
0888      * ENTER IF CMND INTERRUPT WITH FLAG SET AND
0889      * CONTROL SUPPOSEDLY CLEAR.
0890      * -> CONTROL DOES NOT CLEAR OR INT. REQUEST
0891      * STUCK ON!
0892      *
0893      *****
0894      SPC 1

```

```

0895 TST3 EQU *
0896 SSB,RSS IF BIT 15 IS SW-REG CLEAR THEN
0897 HLT 07B ERROR HALT *****
0898 SPC 1
0899 CON2 EQU *
0900 LDA JT4 RESET THE CMND INT.
0901 G.C6 STA CMND LOCATION
0902 SPC 1
0903 G.C7 STC CMND FORCE CMND INTERRUPT
0904 NOP WAIT
0905 NOP WAIT
0906 NOP WAIT
0907 SSB,RSS IF BIT 15 OF SW-REG CLEAR THEN
0908 HLT 10B ERROR HALT *****
0909 SPC 1
0910 *****
0911 *
0912 * ENTER ON EXPECTED CMND INTERRUPT
0913 *
0914 *****
0915 SPC 1
0916 TST4 EQU *
0917 SPC 1
0918 CLC 00,C DISABLE DEVICES & INTERRUPTS
0919 G.D10 CLC DATA,C TURN OFF DATA CHAN.
0920 G.C8 CLC CMND,C TURN OFF CMND CHAN.
0921 SPC 1
0922 SSB IF BIT 15 OF SW-REG SET THEN
0923 JMP TL1 LOOP
0924 SPC 1
0925 HLT 11B SUCCESS HALT *****
0926 SPC 1
0927 LIB SWREG LOAD SW-REG
0928 SSB IF BIT 15 SET THEN
0929 JMP TL1 LOOP
0930 SPC 1
0931 *****
0932 *
0933 * CHECK FOR SINGLE/DOUBLE DENSITY BIT IN STATUS WORD
0934 *
0935 *****
0936 SPC 1
0937 G.C9 LIA CMND LOAD STATUS WORD
0938 LDB A COPY INTO B-REG
0939 AND B40 ISOLATE DOUBLE DENSITY BIT
0940 SWP SWAP A-REG & B-REG
0941 CPB DD IF NOT DENSITY THAT
0942 RSS USER SUPPLIED THEN
0943 HLT 12B ERROR HALT *****
0944 HED FLOPPY DIAGNOSTIC - GET THE FLOPPY UNIT #
0945 *****
0946 *
0947 * SW-REG = FLOPPY UNIT SELECT NUMBER (0-3) IN BITS 1-0
0948 *
0949 * CLEAR BIT 15 TO CONTINUE WITH DIAGNOSTIC.
0950 * SET BIT 15 TO GO TO "INITIALIZATION OF USER FLOPPY"
0951 * SECTION OF DIAGNOSTIC.
0952 *
0953 *****
0954 SPC 1
0955 HLT 13B HALT FOR DATA *****
0956 LIA SWREG LOAD SW-REG
0957 AND D3 ISOLATE UNIT #
0958 RAR,RAR ROTATE UNIT # INTO BITS 15-14

```

```

0959      STA UNITH      SAVE FOR LATER USE
0960      SPC 1
0961      LIA SWREG      LOAD SW-REG
0962      SSA            IF SIGN SET THEN
0963      JMP FINIT      GO TO USER INIT SECTION
0964      HED FLOPPY DIAGNOSTIC - ADDR FIELD TEST
0965      *****
0966      *
0967      * ZERO TRACK & WRITE OUT A SINGLE ADDR FIELD *
0968      * TEST INIT MODE FUNCTIONS *
0969      *
0970      *****
0971      SPC 1
0972  STRT1 EQU *
0973      LDA D10          SET FOR 5 WORDS
0974      STA BYTES
0975      SPC 1
0976      CLA,INA          SELECT TRACK 1
0977      STA TRACK
0978      SPC 1
0979      INA              SELECT SECTOR 2
0980      STA SECTR
0981      SPC 1
0982      LDA BYTES
0983      LDB SECTR
0984      RRL 8
0985      STB SCHOS
0986      SPC 1
0987      LDB TRACK
0988      RRL 8
0989      STB TRLOS
0990      SPC 1
0991      LDA B2K
0992      IOR UNITH
0993      STA CW2
0994      SPC 1
0995      IOR B1004
0996      STA CW3
0997      SPC 1
0998      LDA DM5.2
0999      STA COUNT
1000      SPC 1
1001      LDA B14K
1002      IOR UNITH
1003      IOR TRACK
1004      STA CW0
1005      SPC 1
1006  S.D1  STC DATA,C
1007  S.C1  STC CMND,C
1008      SPC 1
1009  S.C2  OTA CMND      TURN ON INIT MODE
1010      SPC 1
1011      LIA SWREG      LOAD SW-REG
1012      SSA            IF SIGN BIT SET THEN
1013      JMP S1          CONTINUE
1014      SPC 1
1015  S.C3  LIB CMND      GET STATUS
1016      BLF,BLF        SELECT INIT ERROR BIT
1017      SLB             IF SET THEN
1018      HLT 148         ERROR HALT *****
1019      SPC 1
1020      LIA SWREG      LOAD SW-REG
1021      SSA            IF SIGN BIT SET THEN
1022      JMP S1          CONTINUE

```

426

50003

1023		SPC 1	
1024		BLF	SELECT INIT MODE BIT
1025		SLB,RSS	IF NOT SET THEN
1026		HLT 158	ERROR HALT ***** 17 = 4438
1027		SPC 1	
1028	S1	EQU *	
1029	S.C4	SFS CMND	WAIT FOR DEVICE READY
1030		JMP *-1	
1031	S.C5	CLF CMND	CLEAR CMND FLAG
1032		SPC 1	
1033		JMP CLR1A	
1034		SPC 1	
1035	SET1A	EQU *	
1036		CLA	
1037		SPC 1	
1038	SET1	EQU *	
1039	S.C6	LIB CMND	
1040		SSB,RSS	
1041		JMP CLR1A	
1042		SPC 1	
1043		ISZ A	IF NOT TIMED OUT THEN
1044		JMP SET1	LOOP
1045		SPC 1	
1046		LIA SWREG	LOAD SW-REG
1047		SSA,RSS	IF SIGN BIT NOT SET THEN
1048		HLT 168	ERROR HALT *****
1049		SPC 1	
1050	CLR1A	EQU *	
1051		CLA	
1052		SPC 1	
1053	CLR1	EQU *	
1054	S.C7	LIB CMND	
1055		SSB	
1056		JMP BUMP1	
1057		SPC 1	
1058		ISZ A	IF NOT TIMED OUT
1059		JMP CLR1	LOOP
1060		SPC 1	
1061		LIA SWREG	LOAD SW-REG
1062		SSA,RSS	IF SIGN BIT NOT SET THEN
1063		HLT 178	ERROR HALT ***** 471
1064		SPC 1	
1065	BUMP1	EQU *	472/34471
1066		ISZ COUNT	
1067		JMP SET1A	
1068		SPC 1	525
1069		LDA CW2	
1070	S.C8	OTA CMND	474/ 60173
1071		LDA CW3	
1072	S.C9	OTA CMND	
1073		SPC 1	
1074		LDA TRLOS	
1075	S.D2	SFS DATA	
1076		JMP *-1	
1077		SPC 1	
1078	S.D3	OTA DATA,C	503/103615
1079		SPC 1	
1080		LDA SCHOS	
1081	S.D5	SFS DATA	
1082		JMP *-1	
1083		SPC 1	
1084	S.D6	OTA DATA,C	
1085		SPC 1	
1086		LDA CW0	

```

1087 S.C10 SFS CMND
1088     JMP *-1
1089 S.C11 OTA CMND,C
1090     SPC 1
1091 S.C17 SFS CMND      WAIT FOR DEVICE DONE
1092     JMP *-1
1093 S.C18 CLF CMND
1094     SPC 2
1095 *****
1096 *
1097 * READ AN ADDR ID.
1098 *
1099 *****
1100     SPC 1
1101     LDA B24K
1102     IOR UNITH
1103     IOR TRACK
1104     SPC 1
1105 S.D8  STC DATA,C
1106 S.C13 STC CMND,C
1107     SPC 1
1108 S.C14 OTA CMND      824/102616 A. 24004
1109     SPC 1          525
1110 S.D9  SFS DATA
1111     JMP *-1
1112     SPC 1
1113 S.D10 LIA DATA
1114 S.D11 CLF DATA
1115     SPC 1
1116 S.D12 SFS DATA      534
1117     JMP *-1
1118     SPC 1
1119 S.D13 LIB DATA
1120 S.D14 CLF DATA
1121     SPC 1
1122 S.C15 SFS CMND      535
1123     JMP *-1
1124 S.C16 CLF CMND
1125     SPC 1
1126     SPC 1
1127     STB TEMP1
1128     SPC 1
1129     LIB SWREG      LOAD SW-REG
1130     SSB            IF SIGN BIT SET THEN
1131     JMP STRT1      LOOP
1132     SPC 1
1133     LDB TRLOS      LOAD TRACK + LOSS
1134     CPA 8          IF READ NOT = WRITTEN
1135     RSS            THEN
1136     HLT 208        ERROR HALT *****
1137     SPC 1
1138     LDA TEMP1
1139     LDB SCHOS      LOAD SECTOR + HOSS
1140     CPA 8          IF READ NOT = WRITTEN
1141     RSS            THEN
1142     HLT 218        ERROR HALT *****
1143     SPC 1
1144     HLT 228        SUCCESS HALT *****
1145     SPC 1
1146     LIB SWREG      LOAD SW-REG
1147     SSB            IF SIGN BIT SET THEN
1148     JMP STRT1      LOOP
1149     HED FLOPPY DIAGNOSTIC - SECTOR FIELD TEST
1150 *****

```

```

1151 * * *
1152 * ZERO A TRACK. WRITE OUT ADDR ID & SECTOR DATA. *
1153 * READ IN AND CHECK. *
1154 * END WITH A-REG = # WORDS WRITTEN. *
1155 * B-REG = # WORDS READ INCORRECTLY *
1156 * *
1157 *****
1158 SPC 2
1159 CHCK EQU *
1160 LDA DM5.2
1161 STA COUNT
1162 SPC 1
1163 LDA CW0
1164 SPC 1
1165 C.D1 STC DATA,C
1166 C.C1 STC CMND,C
1167 SPC 1
1168 C.C2 OTA CMND
1169 SPC 1
1170 C.C9 SFS CMND WAIT FOR DEVICE READY
1171 JMP *-1
1172 C.C10 CLF CMND
1173 SPC 1
1174 JMP CLR2
1175 SPC 1
1176 SET2 EQU *
1177 C.C3 LIB CMND
1178 SSB
1179 JMP SET2
1180 SPC 1
1181 CLR2 EQU *
1182 C.C4 LIB CMND
1183 SSB,RSS
1184 JMP CLR2
1185 SPC 1
1186 ISZ COUNT
1187 JMP SET2
1188 SPC 1
1189 LDA CW2
1190 C.C5 OTA CMND
1191 LDA CW3
1192 C.C6 OTA CMND
1193 SPC 1
1194 LDA TRLOS
1195 C.D2 SFS DATA
1196 JMP *-1
1197 SPC 1
1198 C.D3 OTA DATA,C
1199 SPC 1
1200 LDA SCHOS
1201 C.D5 SFS DATA
1202 JMP *-1
1203 SPC 1
1204 C.D6 OTA DATA,C
1205 SPC 1
1206 LDA BYTES LOAD # BYTES
1207 ARS DIVIDE BY 2 TO GET WORDS
1208 CMA,INA NEGATE
1209 STA COUNT SAVE IN COUNT
1210 SPC 1
1211 LDA BYTES CONSTRUCT NEW CW2
1212 ALF,ALF
1213 AND B377
1214 IOR B2K

```

1215		IOR UNITH	
1216		STA B	SAVE CW2 IN B-REG
1217		SPC 1	
1218		LDA BYTES	CONSTRUCT NEW CW3
1219		AND B377	
1220		IOR B3K	
1221		IOR UNITH	SAVE CW3 IN A-REG
1222		SPC 1	
1223	C.C7	SFS CMND	
1224		JMP *-1	
1225	C.C8	CLF CMND	
1226		SPC 1	
1227	C.C11	OTB CMND	
1228	C.C12	OTA CMND	
1229		SPC 1	
1230		LDA WORD	
1231	LOOP1	EQU *	
1232	C.D8	SFS DATA	
1233		JMP *-1	
1234		SPC 1	
1235	C.D9	OTA DATA,C	
1236		SPC 1	
1237		ISZ COUNT	
1238		JMP LOOP1	
1239		SPC 1	
1240	C.C13	SFS CMND	
1241		JMP *-1	
1242	C.C14	CLF CMND	
1243		SPC 1	
1244		LDA CW0	
1245	C.C15	OTA CMND	TURN OFF INIT MODE
1246		SPC 1	
1247	C.C21	SFS CMND	WAIT FOR DEVICE END
1248		JMP *-1	
1249	C.C22	CLF CMND	
1250		SKP	
1251	*****		
1252	*		
1253	* TIME TO CHECK DATA		
1254	*		
1255	*****		
1256		SPC 1	
1257		LDA BYTES	LOAD # BYTES
1258		ARS	DIVIDE BY 2 TO GET WORDS
1259		CMA,INA	NEGATE
1260		STA COUNT	SAVE IN COUNT
1261		SPC 1	
1262		LDA B4K	CONSTRUCT CW0 FOR SECTOR READ
1263		IOR UNITH	
1264		IOR TRACK	
1265		LDB A	SAVE CW0 IN B-REG
1266		SPC 1	
1267		LDA B1K	CONSTRUCT CW1
1268		IOR UNITH	
1269		IOR SECTR	SAVE CW1 IN A-REG
1270		SPC 1	
1271	C.D11	STC DATA,C	
1272	C.C16	STC CMND,C	
1273		SPC 1	
1274	C.C17	OTA CMND	OUTPUT CW1
1275	C.C18	OTB CMND	OUTPUT CW0
1276		SPC 1	
1277		CLB	
1278	LOOP2	EQU *	

```

1279 C.D12 SFS DATA
1280     JMP *-1
1281     SPC 1
1282 C.D13 LIA DATA      GET DATA
1283 C.D14 CLF DATA      CLEAR CHANNEL
1284     SPC 1
1285     CPA WORD          IF DATA IS CORRECT THEN
1286     INB               BUMP COUNTER
1287     ISZ COUNT        IF NOT DONE THEN
1288     JMP LOOP2         LOOP
1289     SPC 1
1290 C.C19 SFS CMND
1291     JMP *-1
1292 C.C20 CLF CMND
1293     SPC 1
1294     LIA SWREG         LOAD SW-REG
1295     SSA               IF SIGN SET THEN
1296     JMP CHCK         LOOP
1297     SPC 1
1298     LDA BYTES
1299     ARS
1300     CMB,INB
1301     ADB A
1302     SPC 1
1303     SZB,RSS          IF NO ERRORS THEN
1304     JMP CHCK1        CONTINUE
1305     SPC 1
1306     HLT 238          ERROR HALT *****
1307     SPC 1
1308     RSS
1309 CHCK1 EQU *
1310     HLT 248          SUCCESS HALT *****
1311     SPC 1
1312     LIA SWREG         LOAD SW-REG
1313     SSA               IF SIGN SET THEN
1314     JMP CHCK         LOOP
1315     HED FLOPPY DIAGNOSTIC - INIT. ENTIRE FLOPPY
1316 *****
1317 *
1318 *   INITIALIZE ENTIRE FLOPPY WITH "HARD 6'S" AND
1319 *   READ BACK IN AND CHECK.
1320 *
1321 *   INITIALIZE CONFIGURATION:
1322 *
1323 *   TRACK      # SECTORS/TRACK      # WORDS/SECTOR
1324 *           DD      SD
1325 *   -----
1326 *
1327 *       0           2           2427   1202
1328 *       1           4           1202   589
1329 *       2           8           589    283
1330 *       3          16           283    130
1331 *
1332 *           ETC.
1333 *
1334 *
1335 *   THE DATA WORD WRITTEN IS OCTAL 066666. THIS
1336 *   IS SUPPOSED TO BE AN ENVIGORATING TEST OF THE
1337 *   HARDWARE.
1338 *
1339 *   THE # SECTORS/TRACK WRAPS AROUND TO 2 AGAIN
1340 *   WHEN THE COMPUTED TRACK SIZE IS TOO BIG.
1341 *   FOR DOUBLE DENSITY, THE MAXIMUM IS:
1342 *       128 SECTORS/TRACK, 15 WORDS/SECTOR

```

```

1343 *      WHICH OCCURS EVERY 7-TH TRACK.      *
1344 *      FOR SINGLE DENSITY, THE MAXIMUM IS:  *
1345 *      64 SECTORS/TRACK, 15 WORDS/SECTOR   *
1346 *      WHICH OCCURS EVERY 6-TH TRACK.      *
1347 *                                           *
1348 *      THE ABOVE NUMBERS ASSUME:            *
1349 *      22 WORDS OF OVERHEAD/SECTOR AND      *
1350 *      4900 MAX. WORDS/TRACK (DOUBLE DENSITY), *
1351 *      2450 MAX. WORDS/TRACK (SINGLE DENSITY). *
1352 *                                           *
1353 *****
1354         SPC 2
1355         CLA
1356         STA STRK      STARTING TRACK = 0
1357         STA ETRK      ENDING TRACK = 0
1358         STA SKIP      SKIP COUNT = 0
1359         SPC 1
1360         LDA HSIX
1361         STA WORD      DATA WORD = B066666
1362         SPC 1
1363 UP      EQU *
1364         LDA D2
1365         STA NSEC      # SECTORS/TRACK = 2
1366         SPC 1
1367 YOURS   EQU *
1368         LDB DD        LOAD DOUBLE DENSITY FLAG
1369         LDA NDD        LOAD # WORDS/TRACK: DD
1370         SZB,RSS        IF SINGLE DENSITY THEN
1371         LDA NSD        LOAD # WORDS/TRACK: SD
1372         SPC 1
1373         CLB           CLEAR B-REG FOR DIV
1374         DIV NSEC       DIVIDE BY # SECTORS/TRACK
1375         ADA DM23       SUBTRACT # WORDS OVERHEAD/SECTOR + 1
1376         SPC 1
1377         SZA           IF # WORDS/SECTOR IS ZERO
1378         SSA           OR NEGATIVE THEN
1379         JMP UP        RESET # SECTORS/TRACK
1380         SPC 1
1381         STA SECL       SAVE AS # WORDS/SECTOR
1382         SPC 1
1383 TRYA    EQU *
1384         JSB INIT      INITIALIZE A TRACK
1385         JMP OUT       <INVALID TRACK ADDR>
1386 ***
1387 * IF NSEC/(SKIP+1) NOT AN INTEGER WHEN SKIP = 0
1388 * THEN SOMETHING JUST AIN'T RIGHT....
1389 ***
1390         HLT 74B      ERROR HALT *****
1391 ***
1392 * IF TRACK SIZE IS TOO BIG WHEN I TOOK GREAT PAINS
1393 * TO MAKE SURE IT WASN'T, THEN FORGET YOU....
1394 ***
1395         HLT 74B      ERROR HALT *****
1396         JMP NOSWT    <INIT MODE CAN'T BE ENTERED>
1397         SPC 1
1398 *****
1399 *
1400 *      NORMAL RETURN FROM INITIALIZER.
1401 *      READ BACK IN SECTORS 0 & 1 FOR COMPARISON.
1402 *
1403 *****
1404         SPC 1
1405         LDA STRK
1406         STA TRACK

```

```

1407      SPC 1
1408      LDA SECL      LOAD # WORDS/SECTOR
1409      CMA, INA      NEGATE
1410      STA BUFL      SAVE AS READ LENGTH
1411      SPC 1
1412      CLA          SELECT SECTOR 0
1413      STA SECTR
1414      SPC 1
1415      LDB ABUF      LOAD BUFFER ADDR
1416      JSB .READ, I  READ SECTOR 0
1417      JMP G01      NORMAL RETURN
1418      HLT 25B      <ADDR CHECK> ERROR HALT *****
1419      JMP COFLG     <COMMAND FLAG TIME OUT>
1420      SPC 1
1421  G01      EQU *
1422      JSB .CHCK, I  IF DATA DOES NOT CHECK THEN
1423      HLT 26B      ERROR HALT *****
1424      SPC 1
1425      ISZ SECTR     SELECT SECTOR 1
1426      LDB ABUF      LOAD BUFFER ADDR
1427      JSB .READ, I  READ SECTOR 1
1428      JMP G02      <NORMAL RETURN>
1429      HLT 25B      <ADDR CHECK> ERROR HALT *****
1430      JMP COFLG     <CMND FLAG TIME OUT>
1431      SPC 1
1432  G02      EQU *
1433      JSB .CHCK, I  IF DATA DOES NOT CHECK THEN
1434      HLT 26B      ERROR HALT *****
1435      SPC 1
1436      *****
1437      *
1438      *   CONTINUE TO NEXT TRACK.
1439      *
1440      *****
1441      SPC 1
1442      ISZ STRK      BUMP TRACK COUNTERS
1443      ISZ ETRK
1444      SPC 1
1445      LDA NSEC      LOAD # SECTORS/TRACK
1446      ALS          DOUBLE
1447      STA NSEC      RESAVE
1448      SPC 1
1449      JMP YOURS     LOOP
1450      SPC 1
1451      *****
1452      *
1453      *   INIT MODE CAN'T BE ENTERED.
1454      *   PROBABLY INIT SWITCH NOT UP.
1455      *
1456      *****
1457      SPC 1
1458  NOSWT EQU *
1459      HLT 27B      ERROR HALT *****
1460      JMP TRYA     TRY AGAIN
1461      SPC 1
1462      *****
1463      *
1464      *   DONE WITH ENTIRE FLOPPY INIT TEST
1465      *
1466      *****
1467      SPC 1
1468  OUT      EQU *
1469      HLT 30B      SUCCESSFUL HALT *****
1470      HED FLOPPY DIAGNOSTIC - CRC TEST

```

```

1471 *****
1472 *
1473 * TEST CRC ERROR CHECK.
1474 *
1475 * WRITE OUT A SINGLE 1500 WORD SECTOR ON
1476 * TRACK 76 AND THEN DESTROY THE MIDDLE
1477 * OF IT TO GENERATE A CRC ERROR ON A
1478 * SUBSEQUENT READ.
1479 * CRC MEANS CYCLIC REDUNDENCY CHECK.
1480 *
1481 *****
1482 SPC 1
1483 *****
1484 *
1485 * CURRENTLY, SKIP OVER CRC CHECK DUE TO HARDWARE
1486 * PROBLEMS BEYOND OUR CONTROL.
1487 *
1488 * AS USUSAL.....
1489 *
1490 *****
1491 SPC 1
1492 JMP SKIP1 <SKIP CRC CHECK>
1493 SPC 1
1494 MNGLE EQU *
1495 LDA D76 SELECT TRACK 76
1496 STA STRK
1497 STA ETRK
1498 STA TRACK
1499 SPC 1
1500 LDA HSIX
1501 STA WORD DATA WORD = B066666
1502 SPC 1
1503 CLA
1504 STA SKIP SKIP COUNT = 0
1505 STA SECTR SELECT SECTOR 0
1506 SPC 1
1507 CLA,INA
1508 STA NSEC # SECTORS/TRACK = 1
1509 SPC 1
1510 LDA D1.5
1511 STA SECL # WORDS/SECTOR = 1500
1512 SPC 1
1513 TRYAI EQU *
1514 JSB INIT INITIALIZE TRACK 76
1515 ***
1516 * INVALID TRACK ADDR?!? I'LL INVALID YOU!....
1517 ***
1518 HLT 748 ERROR HALT *****
1519 ***
1520 * NSEC/(SKIP+1) NOT AN INTEGER WHEN SKIP = 0 ?!?
1521 * FMTT....
1522 ***
1523 HLT 748 ERROR HALT *****
1524 ***
1525 * I DON'T CARE WHAT DENSITY YOU ARE, IF YOU CAN'T FIT
1526 * ONE LOUSY 1500 WORD SECTOR, YOU DON'T DESERVE TO RUN...
1527 ***
1528 HLT 748 ERROR HALT *****
1529 ***
1530 * INIT MODE CAN'T BE ENTERED.
1531 * TURKEY FORGOT TO SET UP INIT SWITCH AGAIN.
1532 ***
1533 JMP SWITCH
1534 ***

```

```

1535 * SUCCESSFUL COMPLETION OF INIT.  NOW GO DESTROY CRC.
1536 ***
1537     JMP DESTR
1538     SPC 2
1539 *****
1540 *
1541 *  INIT MODE CAN'T BE ENTERED.  PROBABLY INIT SWITCH DOWN.
1542 *
1543 *****
1544     SPC 1
1545 SWITCH EQU *
1546     HLT 318      ERROR HALT *****
1547     JMP TRYA1     TRY AGAIN
1548     SPC 1
1549 *****
1550 *
1551 *  NOW THAT TRACK 76 IS INITIALIZED, LET'S DESTROY CRC.
1552 *
1553 *****
1554     SPC 1
1555 DESTR EQU *
1556     SPC 1
1557 ***
1558 * READ THE ADDR ID
1559 ***
1560     SPC 1
1561     LDA B24K      CONSTRUCT CWO
1562     IOR UNITH
1563     IOR TRACK
1564     SPC 1
1565 D.D1 STC DATA,C  COCK FLOPPY
1566 D.C1 STC CMND,C
1567     SPC 1
1568 D.C2 OTA CMND     START UP
1569     SPC 1
1570 ***
1571 * READ IN ADDR ID DATA, THEN THROW AWAY.
1572 ***
1573     SPC 1
1574 D.D3 SFS DATA
1575     JMP *-1
1576     SPC 1
1577 D.D4 LIA DATA
1578 D.D5 CLF DATA
1579     SPC 1
1580 D.D6 SFS DATA
1581     JMP *-1
1582     SPC 1
1583 D.D7 LIB DATA
1584 D.D8 CLF DATA
1585     SPC 1
1586 D.C3 SFS CMND     WAIT FOR DEVICE END
1587     JMP *-1
1588 D.C4 CLF CMND
1589     SPC 1
1590 ***
1591 * WAIT FOR ABOUT 4 MSEC, UNTIL WE'RE INTO THE DATA AREA.
1592 ***
1593     SPC 1
1594     LDB DM1K
1595     ISZ B
1596     JMP *-1
1597     SPC 1
1598 ***

```

```

1599 * NOW TURN ON INITIALIZE TO BLOW CRC OFF THE WALL.
1600 ***
1601 SPC 1
1602 LDA B14K CONSTRUCT CWO
1603 IOR UNITH
1604 IOR TRACK
1605 SPC 1
1606 D.D2 STC DATA,C COCK FLOPPY
1607 D.C11 STC CMND,C
1608 SPC 1
1609 D.C5 OTA CMND TURN ON INIT.
1610 SPC 1
1611 D.C6 SFS CMND WAIT FOR DEVICE READY
1612 JMP *-1
1613 D.C7 CLF CMND
1614 SPC 1
1615 ***
1616 * WAIT ABOUT 4 MSEC.
1617 ***
1618 SPC 1
1619 LDB DM1K
1620 ISZ B
1621 JMP *-1
1622 SPC 1
1623 ***
1624 * NOW TURN OFF INITIALIZE.
1625 ***
1626 SPC 1
1627 D.C8 OTA CMND TURN OFF INIT.
1628 SPC 1
1629 D.C9 SFS CMND
1630 JMP *-1
1631 D.C10 CLF CMND
1632 SPC 1
1633 *****
1634 *
1635 * HA-HA.....NOW TRY AND READ THE DATA.
1636 *
1637 *****
1638 SPC 1
1639 LDA SECL LOAD SECTOR LENGTH
1640 CMA,INA NEGATE
1641 STA BUFL SAVE AS READ LENGTH
1642 SPC 1
1643 LDB ABUF LOAD BUFFER ADDR
1644 JSB .READ,I READ SECTOR 0, TRACK 76
1645 JMP AHA! <NORMAL RETURN>
1646 JMP AHA! <ADDR OR DATA CHECK>
1647 JMP COFLG <CMND FLAG TIME OUT>
1648 SPC 1
1649 ***
1650 * NOW CHECK STATUS WORD FOR CRC ERROR BIT.
1651 ***
1652 SPC 1
1653 AHA! EQU *
1654 LDA B LOAD STATUS
1655 AND B20K SELECT DATA CHECK ERROR BIT
1656 SZA IF BIT SET THEN
1657 JMP OUT3 OK-DOKY
1658 SPC 1
1659 HLT 328 ELSE ERROR HALT *****
1660 SPC 1
1661 RSS
1662 OUT3 EQU *

```

```

1663          HLT 338          SUCCESSFUL HALT *****
1664          SPC 1
1665          LDA SWREG          LOAD SW-REG
1666          SSA                IF SIGN BIT SET THEN
1667          JMP MNGLE          LOOP
1668          SPC 2
1669  SKIP1 EQU *
1670          HED FLOPPY DIAGNOSTIC - READ/WRITE TEST
1671  *****
1672  *
1673  *   READ/WRITE HARDWARE TEST
1674  *
1675  *   INITIALIZE TRACK 5 WITH ONE 1000 WORD SECTOR.
1676  *   WRITE ALTERNATING BITS (10101...) THEN READ
1677  *   AND CHECK. ROTATE THE BIT PATTERN (01010...)
1678  *   AND REPEAT.
1679  *
1680  *   DO THIS WHOLE PROCESS 5 TIMES FOR GOOD MEASURE.
1681  *
1682  *****
1683          SPC 1
1684          LDA D5              LOAD DECIMAL 5
1685          STA STRK            SAVE AS STARTING TRACK
1686          STA ETRK            SAVE AS ENDING TRACK
1687          STA TRACK           SAVE AS READ/WRITE TRACK
1688          SPC 1
1689          LDA DM5             LOAD NEGATIVE 5
1690          STA CNT1            SAVE AS LOOP COUNT
1691          SPC 1
1692          LDA D1K              # WORDS/SECTOR = 1000
1693          STA SECL
1694          CMA, INA            RECORD LENGTH =
1695          STA BUFL             -(# WORDS/SECTOR)
1696          SPC 1
1697          CLA, INA            # SECTORS/TRACK = 1
1698          STA NSEC
1699          SPC 1
1700          CLA
1701          STA SECTR            SELECT SECTOR 0
1702          STA SKIP            SKIP COUNT = 0
1703          STA WORD            DATA WORD = 0
1704          SPC 1
1705  TRYA2 EQU *
1706          JSB INIT            INITIALIZE TRACK
1707          HLT 74B             <INVALID TRACK> *****
1708          HLT 74B             <NSEC/(SKIP+1) NOT INTEGER> *****
1709          HLT 74B             <TRK TOO BIG> *****
1710          RSS                 <INIT MODE CAN'T BE ENTERED>
1711          JMP RWOK1           <INIT SUCCESSFUL>
1712          SPC 1
1713  ***
1714  *   INIT MODE CAN'T BE ENTERED.
1715  *   INIT SWITCH PROBABLY UP.
1716  ***
1717          SPC 1
1718          HLT 34B             ERROR HALT *****
1719          JMP TRYA2           TRY AGAIN
1720          SPC 2
1721  *****
1722  *
1723  *   SUCCESSFUL INITIALIZATION.
1724  *   START ALTERNATING BITS TEST OF HARDWARE.
1725  *
1726  *****

```

```

1727         SPC 1
1728 RWOK1 EQU *
1729         LDA ALTB          LOAD ALTERNATING BITS
1730         STA WORD          SAVE AS DATA WORD
1731         SPC 1
1732 RWLP1 EQU *
1733         SPC 1
1734 ***
1735 * ROTATE THE DATA
1736 ***
1737         SPC 1
1738         LDA BUFL
1739         STA TEMP1
1740         SPC 1
1741         LDA WORD          LOAD BITS
1742         CMA              FLIP
1743         STA WORD          REPLACE AS DATA
1744         SPC 1
1745         LDB ABUF          LOAD XFER BUFFER ADDR
1746         SPC 1
1747 RWLP2 EQU *
1748         STA B,I          SAVE BITS IN BUFFER
1749         INB              BUMP BUFFER PTR
1750         ISZ TEMP1        BUMP BUFFER COUNTER
1751         JMP RWLP2        IF NOT DONE THEN LOOP
1752         SPC 1
1753 ***
1754 * WRITE DATA
1755 ***
1756         SPC 1
1757         LDB ABUF          LOAD BUFFER ADDR
1758         JSB .WRIT,I      WRITE SECTOR
1759         JMP RWOK3        <SUCCESSFUL WRITE>
1760         RSS              <ADDR CHECK>
1761         JMP COFLG        <CMD FLAG TIME OUT>
1762         SPC 1
1763 ***
1764 * ADDR CHECK ON WRITE
1765 ***
1766         SPC 1
1767         HLT 378          ERROR HALT *****
1768         SPC 1
1769         LIA SWREG        LOAD SW-REG
1770         SSA,RSS          IF SIGN BIT CLEAR THEN
1771         JMP RWEXT        EXIT TEST
1772         SPC 1
1773 RWOK3 EQU *
1774         SPC 1
1775 ***
1776 * ZERO THE XFER BUFFER FOR READ
1777 ***
1778         SPC 1
1779         LDA BUFL
1780         STA TEMP1
1781         LDB ABUF
1782         CLA
1783         SPC 1
1784 RWLP3 EQU *
1785         STA B,I
1786         INB
1787         ISZ TEMP1
1788         JMP RWLP3
1789         SPC 1
1790 ***

```

```

1791 * READ DATA
1792 ***
1793     LDB ABUF          LOAD BUFFER ADDR
1794     JSB .READ,I      READ SECTOR
1795     JMP RWOK2        <SUCCESSFUL READ>
1796     RSS              <ADDR CHECK>
1797     JMP COFLG        <CMND FLAG TIME OUT>
1798     SPC 1
1799 ***
1800 * ADDR CHECK ON READ.
1801 ***
1802     SPC 1
1803     HLT 35B          ERROR HALT *****
1804     SPC 1
1805     LIA SWREG        LOAD SW-REG
1806     SSA,RSS          IF SIGN CLEAR THEN
1807     JMP RWEXT        EXIT TEST
1808     SPC 1
1809 ***
1810 * CHECK DATA
1811 ***
1812     SPC 1
1813     RWOK2 EQU *
1814     JSB .CHKC,I      IF DATA DOES NOT CHECK THEN
1815     HLT 36B          ERROR HALT *****
1816     SPC 1
1817 ***
1818 * IF NOT DONE WITH TRIES THEN LOOP
1819 ***
1820     SPC 1
1821     ISZ CNT1
1822     JMP RWLP1
1823     SPC 1
1824 ***
1825 * EXIT TEST
1826 ***
1827     SPC 1
1828     RWEXT EQU *
1829     HLT 40B          *****
1830     HED FLOPPY DIAGNOSTIC - FLOPPY INIT. PARAMETER INPUT
1831     FINIT EQU *
1832     SPC 1
1833     HLT 50B          *****
1834     SPC 1
1835 *****
1836 * SW-REG = STARTING TRACK ADDR *
1837 *****
1838     SPC 1
1839     LIB SWREG
1840     STB STRK
1841     HLT 51B          *****
1842     SPC 1
1843 *****
1844 * SW-REG = ENDING TRACK ADDR *
1845 *****
1846     SPC 1
1847     LIB SWREG
1848     STB ETRK
1849     HLT 52B          *****
1850     SPC 1
1851 *****
1852 * SW-REG = # SECTORS/TRACK *
1853 *****
1854     SPC 1

```

```

1855      LIB SWREG
1856      STB NSEC
1857      HLT 538      *****
1858      SPC 1
1859      *****
1860      * SW-REG = # WORDS/SECTOR *
1861      *****
1862      SPC 1
1863      LIB SWREG
1864      STB SECL
1865      HLT 548      *****
1866      SPC 1
1867      *****
1868      * SW-REG = SKIP COUNT FOR SECTOR INTERLACING *
1869      *****
1870      SPC 1
1871      LIB SWREG
1872      STB SKIP
1873      HLT 558      *****
1874      SPC 1
1875      *****
1876      * SW-REG = DATA WORD TO WRITE INTO NEW SECTORS *
1877      *****
1878      SPC 1
1879      LIB SWREG
1880      STB WORD
1881      SPC 2
1882      RESRT EQU *
1883      JSB INIT
1884      JMP ER1      INVALID TRACK ADDR
1885      JMP ER2      NSEC/(SKIP+1) NOT AN INTEGER
1886      JMP ER3      COMPUTED TRACK SIZE TOO BIG
1887      JMP ER4      INIT SWITCH NOT ENGAGED
1888      SPC 1
1889      JMP OK      NO ERROR
1890      SPC 2
1891      ER1 EQU *      INVALID TRACK ADDR
1892      HLT 568      *****
1893      JMP FINIT
1894      SPC 1
1895      ER2 EQU *      NSEC/(SKIP+1) NOT AN INTEGER
1896      HLT 578      *****
1897      JMP FINIT
1898      SPC 1
1899      ER3 EQU *      COMPUTED TRACK SIZE TOO BIG
1900      HLT 608      *****
1901      JMP FINIT
1902      SPC 1
1903      ER4 EQU *      INIT MODE CAN'T BE ENTERED
1904      LDA CTRK      LOAD TRACK ADDR
1905      ALF,ALF      SWAP BYTES
1906      IDR COUNT     INCLUDE SECTOR ADDR
1907      HLT 618      *****
1908      SPC 1
1909      LIB SWREG      LOAD SW-REG
1910      SSB            IF SIGN BIT SET THEN
1911      JMP RESRT      RESTART INIT WITH SAME PARAMETERS
1912      JMP FINIT      ELSE RE-ENTER PARAMETERS
1913      SPC 2
1914      *****
1915      *
1916      *   DONE WITH INITIALIZATION, NOW CHECK DATA TO
1917      *   VERIFY MEDIA.
1918      *

```

```

1919 *****
1920 SPC 1
1921 OK EQU *
1922 LDA STRK LOAD STARTING TRACK ADDR
1923 STA TRACK SAVE AS CURRENT TRACK ADDR
1924 SPC 1
1925 LDA SECL LOAD # WORDS/SECTOR
1926 CHA, INA NEGATE
1927 STA BUFL SAVE AS READ LENGTH
1928 SPC 1
1929 ***
1930 * START A NEW TRACK
1931 ***
1932 SPC 1
1933 RSS SKIP THE TRACK ADDR INCR.
1934 SPC 1
1935 OKTRK EQU *
1936 ISZ TRACK BUMP TRACK ADDR
1937 LDA TRACK LOAD CURRENT TRACK ADDR
1938 CHA, INA NEGATE
1939 ADA ETRK ADD ENDING TRACK ADDR
1940 SSA IF NEGATIVE THEN
1941 JMP OKOUT DONE WITH TRACKS
1942 SPC 1
1943 CLA SELECT SECTOR 0
1944 STA SECTR
1945 SPC 1
1946 ***
1947 * START A NEW SECTOR
1948 ***
1949 SPC 1
1950 OKSEC EQU *
1951 LDA SECTR LOAD SECTOR ADDR
1952 CPA NSEC IF EQUALS # SECTORS/TRACK THEN
1953 JMP OKTRK START A NEW TRACK
1954 SPC 1
1955 LDB ABUF LOAD BUFFER ADDR
1956 JSB .READ, I READ SECTOR
1957 JMP OKOK1 <SUCCESSFUL READ>
1958 JMP OKUH <ADDR CHECK>
1959 JMP COFLG <CMND FLAG TIMED OUT>
1960 SPC 1
1961 ***
1962 * SUCCESSFUL SECTOR READ
1963 ***
1964 SPC 1
1965 OKOK1 EQU *
1966 JSB .CHK, I IF DATA DOES NOT CHECK THEN
1967 RSS SKIP THE NEXT INSTRUCTION
1968 JMP OKOK2 <DATA MATCHES>
1969 SPC 1
1970 HLT 638 ERROR HALT *****
1971 RSS SKIP THE NEXT INSTRUCTION
1972 SPC 1
1973 OKUH EQU * [ADDR CHECK FROM ABOVE READ]
1974 HLT 628 ERROR HALT *****
1975 SPC 1
1976 LIA SWREG LOAD SW-REG
1977 SSA IF SIGN BIT SET THEN
1978 JMP OKOUT EXIT
1979 SPC 1
1980 OKOK2 EQU *
1981 ISZ SECTR BUMP SECTOR ADDR
1982 JMP OKSEC LOOP

```

```

1983          SPC 1
1984      ***
1985      * DONE WITH DATA CHECK
1986      ***
1987          SPC 1
1988      OKOUT EQU *
1989          HLT 648          SUCCESSFUL HALT *****
1990          SPC 1
1991          LIB SMREG        LOAD SW-REG
1992          SSB             IF SIGN BIT SET THEN
1993          JMP RESRT        RESTART INIT WITH SAME PARMS
1994          SPC 1
1995          JMP FINIT        ELSE RE-ENTER PARMS
1996          HED FLOPPY DIAGNOSTIC - DEVICE TIME OUT TEST
1997      *****
1998      *
1999      * TIME OUT FLAG TEST
2000      *
2001      * INCREMENT A PRESET TIME OUT COUNTER (32 BITS)
2002      * AND GIVE AN ERROR RETURN WHEN THE COUNTER
2003      * REACHES ZERO.
2004      *
2005      * INITIALIZE THE PARAMETERS:
2006      *
2007      * CLOB = LOW ORDER BITS OF COUNTER
2008      * CHOB = HIGH ORDER BITS OF COUNTER
2009      *
2010      * CALLING SEQUENCE:
2011      *
2012      * JSB TO
2013      * <NORMAL RETURN>
2014      * <TIME OUT ERROR RETURN>
2015      *
2016      *****
2017          SPC 2
2018      TO      NOP
2019          ISZ CLOB          BUMP COUNTER LOW BITS
2020          JMP TO,I          IF NOT ZERO THEN RETURN
2021          SPC 1
2022          ISZ CHOB          BUMP COUNTER HIGH BITS
2023          JMP TO,I          IF NOT ZERO THEN RETURN
2024          ISZ TO            SET FOR ERROR RETURN
2025          JMP TO,I          GIVE TIME OUT ERROR
2026          SPC 2
2027      CLOB DEF *-*          TO COUNTER LOW ORDER BITS
2028      CHOB DEF *-*          TO COUNTER HIGH ORDER BITS
2029          HED FLOPPY DIAGNOSTIC - FLOPPY INITIALIZATION ROUTINE
2030      *****
2031      *
2032      * FLOPPY INITIALIZATION SUBROUTINE
2033      *
2034      * INITIALIZES FROM A STARTING TRACK TO AN ENDING
2035      * TRACK (INCLUSIVE) WITH A FIXED # SECTORS/TRACK
2036      * AND FIXED # WORDS/SECTOR.  ALLOWS INTERLACING
2037      * OF LOGICALLY (NUMERICALLY) SEQUENTIAL SECTORS.
2038      * FILLS ALL NEW SECTORS WITH A SINGLE DATA WORD
2039      * REPEATED TO FILL SECTOR.
2040      * WORKS FOR DOUBLE OR SINGLE DENSITY.
2041      *
2042      * CALLING SEQUENCE:
2043      *
2044      * UNITH = FLOPPY UNIT # IN BITS 15-14
2045      * DD    = NON-ZERO IF DOUBLE DENSITY
2046      * STRK  = STARTING TRACK ADDR

```

```

2047 *   ETRK   = ENDING TRACK ADDR      *
2048 *   SECL   = # WORDS/SECTOR        *
2049 *   NSEC    = # SECTORS/TRACK      *
2050 *   SKIP    = # OF SECTORS BETWEEN LOGICALLY *
2051 *             (NUMERICALLY) SEQUENTIAL SECTORS *
2052 *   WORD    = DATA WORD TO FILL SECTORS WITH *
2053 * * *
2054 *   JSB INIT
2055 *   <ERROR RETURN: INVALID TRACK ADDR>
2056 *   <ERROR RETURN: NSEC/(SKIP+1) NOT AN INTEGER>
2057 *   <ERROR RETURN: COMPUTED TRACK SIZE TOO BIG>
2058 *   <ERROR RETURN: INIT MODE CANNOT BE ENGAGED.
2059 *             INIT SWITCH ON PANEL NOT ENGAGED,
2060 *             OR INIT MODE BIT IN STATUS WORD
2061 *             WILL NOT SET.
2062 *             'CTRK' CONTAINS THE ADDRESS OF
2063 *             THE LAST TRACK BEING INITIALIZED>
2064 *   <NORMAL RETURN: INITIALIZATION COMPLETED>
2065 *
2066 *   IN ALL RETURN CASES THE INTERRUPT SYSTEM IS OFF.
2067 *
2068 *****
2069     SPC 2
2070 INIT  NOP
2071     SPC 1
2072     LDA STRK      IF STRK < 0
2073     SSA          THEN
2074     JMP INIT,I    ERROR
2075     SPC 1
2076     LDB ETRK      IF ETRK < 0
2077     SSB          THEN
2078     JMP INIT,I    ERROR
2079     SPC 1
2080     CMB,INB       IF ETRK > 76
2081     ADB D76       THEN
2082     SSB          ERROR
2083     JMP INIT,I
2084     SPC 1
2085     CMA,INA       IF STRK > ETRK
2086     ADA ETRK      THEN
2087     SSA          ERROR
2088     JMP INIT,I
2089     SPC 1
2090     ISZ INIT      BUMP RETURN ADDR
2091     SPC 1
2092     LDA SKIP      LOAD THE SKIP COUNT
2093     INA          INCREMENT IT
2094     STA COUNT     AND STORE TEMPORARILY
2095     SPC 1
2096     CLB          CLEAR B-REG FOR DIV
2097     LDA NSEC      NUMERATOR = #SECTORS/TRACK
2098     DIV COUNT     DENOMINATOR = SKIP + 1
2099     STA P         STORE QUOTIENT IN P
2100     SPC 1
2101     SZB          IF NOT AN INTEGER
2102     JMP INIT,I    THEN ERROR EXIT
2103     ISZ INIT      BUMP ERROR RETURN
2104     SPC 1
2105     LDB DD        LOAD DOUBLE DENSITY FLAG
2106     LDA HDD       LOAD #WORDS/TRACK FOR DD
2107     SZB,RSS       IF SINGLE DENSITY THEN
2108     LDA MSD       LOAD #WORDS/TRACK FOR SD
2109     SPC 1
2110     CLB          CLEAR B-REG FOR DIV

```

```

2111      DIV NSEC      QUOTIENT=#WORDS AVAILABLE/SECTOR
2112      CMA,INA      NEGATE IT
2113      ADA SECL      ADD #WORDS DESIRED/SECTOR
2114      ADA D22      ADD #WORDS OVERHEAD/SECTOR
2115      STA SECOV     SAVE -(#EXTRA WORDS/SECTOR)
2116      SPC 1
2117      SSA,RSS      IF >= 0 THEN
2118      JMP INIT,I    ERROR EXIT
2119      ISZ INIT      BUMP RETURN ADDR
2120      SPC 1
2121      *****
2122      * SET UP FLOPPY COMMAND WORDS 2 & 3 *
2123      * FOR ADDR FIELD & SECTOR AREA *
2124      *****
2125      SPC 1
2126      LDA B2K      SETUP CW2 (MSBC)
2127      IOR UNITH    FOR ADDR FIELD
2128      STA A.CW2     AND SAVE
2129      SPC 1
2130      LDA SECL      LOAD # WORDS/SECTOR
2131      ALS          CHANGE TO # BYTES/SECTOR
2132      ALF,ALF      ISOLATE MSBC IN LOW BYTE
2133      AND B377     MASK
2134      IOR A.CW2     INCLUDE CW2 (MSBC)
2135      STA S.CW2     SAVE AS SECTOR AREA CW2
2136      SPC 1
2137      LDA B3K      SET UP CW3 (LSBC)
2138      IOR UNITH    FOR ADDR FIELD
2139      LDB A
2140      IOR D4        (4 BYTES)
2141      STA A.CW3     AND SAVE
2142      SPC 1
2143      LDA SECL      LOAD #WORDS/SECTOR
2144      ALS          CHANGE TO # BYTES/SECTOR
2145      AND B377     ISOLATE LSBC IN LOW BYTE
2146      IOR B         INCLUDE PREVIOUS EFFORTS
2147      STA S.CW3     TO FORM SECTOR AREA CW3
2148      SPC 1
2149      *****
2150      * TURN OFF INTERRUPTS *
2151      *****
2152      SPC 1
2153      CLF 00        DISABLE INTERRUPTS
2154      SPC 1
2155      *****
2156      * START INITIALIZATION PROCESS FOR A TRACK *
2157      *****
2158      SPC 1
2159      LDA STRK      LOAD STARTING TRACK
2160      STA CTRK      SAVE AS CURRENT TRACK
2161      SPC 1
2162      NEWTR EQU *
2163      SPC 1
2164      CLB          ZERO THE
2165      STB COUNT     SECTOR ADDR COUNTER
2166      STB SEED      AND SECTOR ADDR SEED
2167      SPC 1
2168      I.D1 STC DATA,C COCK FLOPPY?
2169      I.C1 STC CMND,C  STROKE GENTLY...
2170      SPC 1
2171      LDA CTRK      SET UP CW0 TO
2172      AND B177     START
2173      IOR B14K     INITIALIZATION
2174      IOR UNITH    PROCESS

```

```

2175      STA CWO          AND SAVE IT
2176      SPC 1
2177      LDB DM2          SET RETRY COUNTER
2178      STB TEMP1        FOR 2 RETRIES
2179      SPC 1
2180  AGAIN EQU *
2181  I.C2  OTA CMND        OUTPUT CWO TO START
2182      SPC 1
2183  I.C3  LIB CMND        GET STATUS
2184      SPC 1
2185      BLF,BLF          SELECT UNIT ERROR BIT
2186      SLB              IF SET THEN
2187      JMP INIT,I        ERROR EXIT
2188      SPC 1
2189      BLF              SELECT INIT MODE BIT
2190      SLB              IF ON THEN
2191      JMP CON            CONTINUE
2192      SPC 1
2193      ISZ TEMP1        ELSE IF THIS IS FIRST TRY
2194      JMP AGAIN          THEN TRY AGAIN
2195      JMP INIT,I        ELSE ERROR EXIT
2196      SPC 1
2197  CON   EQU *
2198  I.C15 SFS CMND        WAIT FOR DEVICE READY
2199      JMP *-1
2200  I.C16 CLF CMND        CLEAR CMND FLAG
2201      SPC 1
2202  *****
2203  * ZERO THE ENTIRE TRACK *
2204  *****
2205      SPC 1
2206      LDA DM5.5        COUNT 5500
2207      STA ZCNTN
2208      JSB ZCNT          ZEROS
2209      SPC 1
2210  *****
2211  * START INITIALIZATION PROCESS FOR A SECTOR *
2212  *****
2213      SPC 1
2214      LDA SECOV        SET UP TO LEAVE A GAP =
2215      STA ZCNTN        # EXTRA WORDS/SECTOR
2216      SPC 1
2217      CLB              LOAD THE SECTOR ADDR SEED
2218      SPC 1
2219  NEWSO EQU *
2220      CPB P            IF SEED = P THEN
2221      JMP EXIT          DONE WITH TRACK: EXIT
2222      SPC 1
2223      LDA A.CW2        LOAD ADDR CW2 FOR LATER USE
2224      SPC 1
2225      JSB ZCNT          LEAVE A GAP
2226      SPC 1
2227  *****
2228  * WRITE OUT ADDR FIELD *
2229  *****
2230      SPC 1
2231  I.C4  OTA CMND        OUTPUT ADDR FIELD CW2
2232      LDA A.CW3        AND
2233  I.C5  OTA CMND,C      CW3
2234      SPC 1
2235      LDB SECOV        SET UP TO LEAVE A GAP =
2236      STB ZCNTN        # EXTRA WORDS/SECTOR
2237      SPC 1
2238      LDA COUNT        LOAD SECTOR ADDR

```

2239		LDB SECL	LOAD SECTOR LENGTH IN WORDS
2240		BLS	CHANGE TO BYTES
2241		RRL 8	ROTATE TO FORM
2242		STA TEMP1	SECTOR + HOSS
2243		SPC 1	
2244		LDA CTRK	LOAD TRACK ADDR
2245		RRL 8	ROTATE TO FORM TRACK + LOSS
2246		SPC 1	
2247		LDB TEMP1	RELOAD SECTOR + HOSS
2248		SPC 1	
2249	IL1	EQU *	
2250	I.02	SFS DATA	IF DATA FLAG CLEAR THEN
2251		JMP IL1	LOOP
2252		SPC 1	
2253	I.03	OTA DATA,C	OUTPUT TRACK + LOSS
2254		SPC 1	
2255	IL2	EQU *	
2256	I.05	SFS DATA	IF DATA FLAG CLEAR THEN
2257		JMP IL2	LOOP
2258		SPC 1	
2259	I.06	OTB DATA,C	OUTPUT SECTOR + HOSS
2260		SPC 1	
2261		*****	
2262		* WRITE OUT SECTOR DATA *	
2263		*****	
2264		SPC 1	
2265		LDA S.CW2	LOAD SECTOR AREA CW2
2266		LDB S.CW3	LOAD SECTOR AREA CW3
2267		SPC 1	
2268	IL3	EQU *	
2269	I.06	SFS CMND	IF CMND FLAG CLEAR THEN
2270		JMP IL3	LOOP
2271		SPC 1	
2272	I.07	OTA CMND	OUTPUT CW2
2273		SPC 1	
2274	I.010	OTB CMND,C	OUTPUT CW3
2275		SPC 1	
2276		LDA COUNT	SECTOR ADDR =
2277		ADA P	SECTOR ADDR + P
2278		STA COUNT	
2279		SPC 1	
2280		LDA NSEC	LOAD # SECTORS/TRACK
2281		CMA,INA	NEGATE IT
2282		ADA COUNT	ADD THE SECTOR ADDR
2283		LDB SEED	ALSO LOAD SECTOR ADDR SEED
2284		INS	AND INCREMENT IT
2285		SPC 1	
2286		SSA	IF SECTOR ADDR >=
2287		JMP THERE	# SECTORS/TRACK THEN
2288		STB SEED	SET COUNT = SEED =
2289		STB COUNT	SEED + 1
2290	THERE	EQU *	
2291		SPC 1	
2292		LDA SECL	LOAD # WORDS/SECTOR
2293		CMA,INA	NEGATE IT
2294		STA TEMP1	SAVE AS OUTPUT COUNTER
2295		SPC 1	
2296		LDA WORD	LOAD THE SECTOR DATA
2297		SPC 1	
2298	IL4	EQU *	
2299	I.08	SFS DATA	IF DATA FLAG CLEAR THEN
2300		JMP IL4	LOOP
2301		SPC 1	
2302	I.09	OTA DATA,C	OUTPUT SECTOR DATA

```

2303      SPC 1
2304      ISZ TEMP1      BUMP OUTPUT COUNT
2305      JMP IL4        IF NOT DONE THEN LOOP
2306      SPC 1
2307      *****
2308      *
2309      * LOAD CM0 FOR POSSIBLE USE AT TRACK END.
2310      * LOAD SECTOR ADDR SEED FOR LATER COMPARISON
2311      * AT TOP OF LOOP.
2312      *
2313      *****
2314      SPC 1
2315      LDA CM0
2316      LDB SEED
2317      SPC 1
2318      IL5 EQU *
2319      I.C12 SFS CMND      IF CMND FLAG CLEAR THEN
2320      JMP IL5            LOOP
2321      SPC 1
2322      I.C13 CLF CMND      CLEAR CMND CHANNEL
2323      SPC 1
2324      JMP NEWSC          LOOP FOR A NEW SECTOR
2325      SPC 2
2326      *****
2327      * DONE WITH TRACK: *
2328      * TURN OFF INITIALIZATION PROCESS AND *
2329      * CHECK FOR END OF INITIALIZE REQUEST *
2330      *****
2331      SPC 1
2332      EXIT EQU *
2333      I.C14 OTA CMND      TURN INITIALIZE OFF
2334      SPC 1
2335      I.C17 SFS CMND      WAIT FOR DEVICE READY
2336      JMP *-1
2337      SPC 1
2338      I.C18 CLF CMND      CLEAR CMND FLAG
2339      SPC 1
2340      LDA CTRK          LOAD CURRENT TRACK ADDR
2341      ISZ CTRK          INCREMENT SAVE AS NEW TRACK ADDR
2342      SPC 1
2343      CPA ETRK          IF OLD TRACK ADDR IS NOT
2344      RSS              EQUAL TO END TRACK ADDR
2345      JMP NEWTR          THEN LOOP
2346      SPC 2
2347      *****
2348      * DONE WITH INITIALIZATION *
2349      *****
2350      SPC 1
2351      ISZ INIT          BUMP RETURN ADDR
2352      I.D12 CLC DATA,C  TURN OFF BOTH
2353      I.C9  CLC CMND,C   DEVICES
2354      JMP INIT,I        EXIT
2355      SKP
2356      *****
2357      * *
2358      * COUNT ZEROS DURING INITIALIZATION *
2359      * *
2360      * CALLING SEQUENCE: *
2361      * *
2362      * ZCNTN = -(# ZEROS TO COUNT) *
2363      * *
2364      * JSB ZCNT *
2365      * *
2366      * THE A-REG IS LEFT INTACT *

```

```

2367 *
2368 *****
2369 SPC 2
2370 ZCNT NOP
2371 SPC 1
2372 CLEAR EQU *
2373 Z.C1 LIB CMND GET STATUS
2374 SSB,RSS IF BIT 15 CLEAR
2375 JMP CLEAR THEN LOOP
2376 SPC 1
2377 ISZ ZCNTN ELSE BUMP COUNTER
2378 RSS IF DONE THEN
2379 JMP ZCNT,I EXIT
2380 SPC 1
2381 SET EQU *
2382 Z.C2 LIB CMND GET STATUS
2383 SSB IF BIT 15 SET THEN
2384 JMP SET LOOP
2385 SPC 1
2386 JMP CLEAR ELSE WAIT FOR SET
2387 SPC 2
2388 ZCNTN DEF *-* -(# ZEROS TO COUNT)
2389 SKP
2390 *****
2391 * INITIALIZATION ROUTINE DATA AREA *
2392 *****
2393 SPC 2
2394 NDD DEC 4900 #WORDS/TRACK: DOUBLE DENSITY
2395 NSD DEC 2450 #WORDS/TRACK: SINGLE DENSITY
2396 SPC 1
2397 P DEF *-* NSEC/(SKIP + 1)
2398 SEED DEF *-* SECTOR ADDR SEED
2399 SECOV DEF *-* -(EXTRA WORDS/SECTOR)
2400 SPC 1
2401 CW0 DEF *-* COMMAND WORD 0
2402 A.CW2 DEF *-* CW2 FOR ADDR FIELD
2403 A.CW3 DEF *-* CW3 FOR ADDR FIELD
2404 S.CW2 DEF *-* CW2 FOR SECTOR AREA
2405 S.CW3 DEF *-* CW3 FOR SECTOR AREA
2406 HED FLOPPY DIAGNOSTIC - CMND FLAG TIME OUT
2407 *****
2408 * CMND FLAG TIMED OUT *
2409 *****
2410 SPC 1
2411 COFLG EQU *
2412 B.C2 LIB CMND GET STATUS
2413 LDA CTRK GET TRACK ADDR
2414 ALF,ALF SWAP BYTES
2415 IOR COUNT INCLUDE SECTOR
2416 HLT 73B *****
2417 JMP *-1 HARD HALT
2418 SPC 2
2419 *****
2420 *
2421 * CROSS PAGE BOUNDARY *
2422 * ACCESS ALL SUBROUTINES AND TABLES *
2423 * USING INDIRECT LINKS *
2424 *
2425 *****
2426 SPC 1
2427 IFN ABSOLUTE
2428 ORG 2000B
2429 XIF END ABSOLUTE
2430 HED FLOPPY DIAGNOSTIC - READ/WRITE ROUTINE

```

```

2431 *****
2432 *
2433 *      FLOPPY DISC READ/WRITE ROUTINES
2434 *
2435 *      CALLING SEQUENCE:
2436 *
2437 *      TRACK = TRACK ADDRESS
2438 *      SECTR = SECTOR ADDRESS
2439 *      UNITH = FLOPPY UNIT NUMBER BITS(15-14)
2440 *      BUFL = -(TRANSFER LENGTH) IE PHYSICAL SECTOR LENGTH
2441 *      B-REG = TRANSFER BUFFER ADDRESS
2442 *
2443 *      JSB READ
2444 *      --OR--
2445 *      JSB WRITE
2446 *
2447 *      <SUCCESSFUL RETURN>
2448 *      <ERROR RETURN: RETRY COUNT EXHAUSTED>
2449 *      <ERROR RETURN: TIME OUT ON FLOPPY CMND FLAG>
2450 *
2451 *      IN ALL CASES:
2452 *
2453 *      B-REG = FLOPPY STATUS WORD
2454 *
2455 *      FOR ERROR RETURNS:
2456 *
2457 *      A-REG = TRACK ADDR IN BITS 15-8,
2458 *              SECTOR ADDR IN BITS 7-0
2459 *
2460 *****
2461      SPC 2
2462 READ  NOP
2463      LDA READ      RESET RETURN ADDRESS
2464      STA WRITE     FOR CONVENIENCE
2465      SPC 1
2466      LDA BUFL      SET DMA CW3
2467      STA D.CW3
2468      SPC 1
2469      CCE
2470      RBL,ERB      SET DMA CW2
2471      LDA B4K      SELECT FLOPPY READ (CW0)
2472      JMP XFER     START TRANSFER
2473      SPC 2
2474 WRITE NOP
2475      CCA          SET DMA CW3
2476      ADA BUFL     (INCLUDING EXTRA WORD)
2477      STA D.CW3
2478      SPC 1
2479      RBL,CLE,ERB  SET DMA CW2
2480      LDA B30K     SELECT FLOPPY WRITE (CW0)
2481      SPC 1
2482 XFER  EQU *
2483      STB D.CW2    SAVE DMA CW2
2484      IOR UNITH    MERGE UNIT W/FLOPPY CW0
2485      SPC 1
2486      IOR TRACK    PUT TRACK INTO FLOPPY CW0
2487      STA F.CW0    SAVE FOR USE
2488      SPC 1
2489      LDA B1K      CREATE FLOPPY CW1
2490      IOR UNITH    MERGE UNIT
2491      IOR SECTR    MERGE SECTOR WITH CW1
2492      STA F.CW1    STORE FOR LATER USE
2493      SPC 1
2494      LDA DM10     SET FOR 10 RETRIES

```

2495		STA RETRY	ON ERROR
2496		SPC 1	
2497	RLOOP	EQU *	
2498	X.D1	STC DATA,C	COCK THE FLOPPY
2499	X.C1	STC CMND,C	STROKE GENTLY
2500		LDA D.CW1	GET DMA CONTROL WORD
2501		OTA 6	SET UP DMA 2&6
2502		CLC 2	
2503		LDA D.CW2	
2504		OTA 2	
2505		LDA D.CW3	
2506		STC 2	
2507		OTA 2	
2508		STC 6,C	START DMA
2509		CLC 6	INHIBIT DMA INTERRUPT
2510		JSB FSTRT	START FLOPPY, WAIT FOR END
2511		STF 6	FORCE DMA COMPLETION
2512		SPC 2	
2513	RCONT	EQU *	
2514	X.C3	LIA CMND	GET STATUS
2515		LDB A	COPY INTO B-REG
2516		AND B37K3	IGNORE SOME BITS
2517		SZA,RSS	SKIP ON ERROR
2518		JMP WRITE,I	OK? RETURN THEN
2519		SPC 1	
2520		ISZ RETRY	RETRIED 10 TIMES?
2521		JMP R	NO-TRY ONCE MORE
2522		SPC 1	
2523	RR	EQU *	
2524		ISZ WRITE	YES--SOLID ERROR: BUMP RETURN
2525		LDA TRACK	LOAD TRACK ADDR
2526		ALF,ALF	SWAP BYTES
2527		IOR SECTR	INCLUDE SECTOR ADDR
2528		JMP WRITE,I	ERROR EXIT
2529		SPC 1	
2530	R	LDA F.CW0	DO REZERO/SEEK
2531		STA TEMP1	
2532		IOR B34K	
2533		STA F.CW0	
2534		JSB FSTRT	
2535		LDA TEMP1	ATTEMPT OPERATION
2536		STA F.CW0	AGAIN
2537		JMP RLOOP	
2538		SPC 2	
2539	X.D2	ABS 20000B+DATA	
2540	D.CW1	EQU X.D2	
2541	D.CW2	DEF **	
2542	D.CW3	DEF **	
2543	F.CW0	DEF **	
2544	F.CW1	DEF **	
2545	RETRY	DEF **	RETRY COUNTER
2546		SPC 2	
2547	*		
2548	*		
2549	*	START FLOPPY I/O AND AWAIT COMPLETION	
2550	*		
2551	*		
2552	FSTRT	NOP	
2553		LDA F.CW1	OUTPUT FLOPPY
2554	F.C1	OTA CMND	COMMAND WORD
2555		SPC 1	
2556		LDA F.CW0	SECOND WORD
2557	F.C2	OTA CMND	
2558		SPC 1	

```

2559      CLA                SET UP THE
2560      STA CLOB            TIME OUT
2561      LDA DM10           COUNTER
2562      STA CHOB
2563      SPC 1
2564      FL1 EQU *
2565      F.C3 SFC CMND        IF CMND FLAG SET
2566      JMP FSTRT,I        THEN EXIT
2567      SPC 1
2568      JSB TO              ELSE IF NO TIME OUT THEN
2569      JMP FL1             LOOP
2570      SPC 1
2571      F.C4 LDR CMND        ELSE GET STATUS
2572      ISZ WRITE           BUMP RETURN ADDR
2573      JMP RR              ERROR EXIT
2574      HED FLOPPY DIAGNOSTIC - BUFFER CONTENTS CHECK
2575      *****
2576      *
2577      * COMPARE DATA BUFFER AGAINST DATA WORD
2578      *
2579      * CALLING SEQUENCE:
2580      *
2581      * WORD = DATA WORD
2582      * ABUF = ADDR OF BUFFER TO COMPARE
2583      * BUFL = -(BUFFER LENGTH)
2584      * TRACK = TRACK ADDR (FOR ERROR RETURN)
2585      * SECTR = SECTOR ADDR (FOR ERROR RETURN)
2586      *
2587      * JSB CHECK
2588      * <ERROR RETURN: MATCH FAILED.
2589      * B-REG = BUFFER WORD WHICH DOES
2590      * NOT MATCH.
2591      * A-REG = TRACK ADDR IN BITS 15-8
2592      * SECTOR ADDR IN BITS 7-0>
2593      * <SUCCESSFUL RETURN: ENTIRE BUFFER MATCHES DATA WORD>
2594      *
2595      *****
2596      SPC 1
2597      CHECK NOP
2598      LDA ABUF            LOAD BUFFER ADDR
2599      STA TEMP1          SAVE AS PTR
2600      SPC 1
2601      LDA BUFL            -(LOAD BUFFER LENGTH)
2602      SPC 1
2603      CHEC1 EQU *
2604      LDB TEMP1,I
2605      CPB WORD
2606      RSS
2607      JMP BAD
2608      SPC 1
2609      ISZ TEMP1          BUMP BUFFER PTR
2610      ISZ A              BUMP COUNTER, IF NOT DONE
2611      JMP CHEC1          THEN LOOP
2612      SPC 1
2613      ISZ CHECK          BUMP FOR
2614      JMP CHECK,I        GOOD EXIT
2615      SPC 1
2616      BAD EQU *
2617      LDA TRACK          LOAD TRACK ADDR
2618      ALF,ALF            SWAP BYTES
2619      IOR SECTR          INCLUDE SECTOR ADDR
2620      JMP CHECK,I        ERROR EXIT
2621      HED FLOPPY DIAGNOSTIC - TABLES
2622      *****

```

```

2623 * CMND CHANNEL I/O INSTRUCTIONS *
2624 *****
2625 SPC 1
2626 CTB EQU *
2627 DEF B.C2
2628 DEF C.C1
2629 DEF C.C2
2630 DEF C.C3
2631 DEF C.C4
2632 DEF C.C5
2633 DEF C.C6
2634 DEF C.C7
2635 DEF C.C8
2636 DEF C.C9
2637 DEF C.C10
2638 DEF C.C11
2639 DEF C.C12
2640 DEF C.C13
2641 DEF C.C14
2642 DEF C.C15
2643 DEF C.C16
2644 DEF C.C17
2645 DEF C.C18
2646 DEF C.C19
2647 DEF C.C20
2648 DEF C.C21
2649 DEF C.C22
2650 DEF D.C1
2651 DEF D.C2
2652 DEF D.C3
2653 DEF D.C4
2654 DEF D.C5
2655 DEF D.C6
2656 DEF D.C7
2657 DEF D.C8
2658 DEF D.C9
2659 DEF D.C10
2660 DEF D.C11
2661 DEF F.C1
2662 DEF F.C2
2663 DEF F.C3
2664 DEF F.C4
2665 DEF G.C1
2666 DEF G.C2
2667 DEF G.C3
2668 DEF G.C4
2669 DEF G.C5
2670 DEF G.C6
2671 DEF G.C7
2672 DEF G.C8
2673 DEF G.C9
2674 DEF G.C10
2675 DEF I.C1
2676 DEF I.C2
2677 DEF I.C3
2678 DEF I.C4
2679 DEF I.C5
2680 DEF I.C6
2681 DEF I.C7
2682 DEF I.C9
2683 DEF I.C10
2684 DEF I.C12
2685 DEF I.C13
2686 DEF I.C14

```

```

2687      DEF I.C15
2688      DEF I.C16
2689      DEF I.C17
2690      DEF I.C18
2691      DEF S.C1
2692      DEF S.C2
2693      DEF S.C3
2694      DEF S.C4
2695      DEF S.C5
2696      DEF S.C6
2697      DEF S.C7
2698      DEF S.C8
2699      DEF S.C9
2700      DEF S.C10
2701      DEF S.C11
2702      DEF S.C13
2703      DEF S.C14
2704      DEF S.C15
2705      DEF S.C16
2706      DEF S.C17
2707      DEF S.C18
2708      DEF X.C1
2709      DEF X.C3
2710      DEF Z.C1
2711      DEF Z.C2
2712      DEF *+1
2713      OCT 0
2714      SPC 2
2715      *****
2716      * DATA CHANNEL I/O INSTRUCTIONS *
2717      *****
2718      SPC 1
2719      DTB EQU *
2720      DEF C.D1
2721      DEF C.D2
2722      DEF C.D3
2723      DEF C.D5
2724      DEF C.D6
2725      DEF C.D8
2726      DEF C.D9
2727      DEF C.D11
2728      DEF C.D12
2729      DEF C.D13
2730      DEF C.D14
2731      DEF D.D1
2732      DEF D.D2
2733      DEF D.D3
2734      DEF D.D4
2735      DEF D.D5
2736      DEF D.D6
2737      DEF D.D7
2738      DEF D.D8
2739      DEF G.D1
2740      DEF G.D2
2741      DEF G.D3
2742      DEF G.D4
2743      DEF G.D5
2744      DEF G.D6
2745      DEF G.D7
2746      DEF G.D8
2747      DEF G.D9
2748      DEF G.D10
2749      DEF I.D1
2750      DEF I.D2

```

END
OF TABLE

```

2751      DEF I.D3
2752      DEF I.D5
2753      DEF I.D6
2754      DEF I.D8
2755      DEF I.D9
2756      DEF I.D12
2757      DEF S.D1
2758      DEF S.D2
2759      DEF S.D3
2760      DEF S.D5
2761      DEF S.D6
2762      DEF S.D8
2763      DEF S.D9
2764      DEF S.D10
2765      DEF S.D11
2766      DEF S.D12
2767      DEF S.D13
2768      DEF S.D14
2769      DEF X.D1
2770      DEF X.D2
2771      DEF *+1      END
2772      OCT 0      OF TABLE
2773      HED FLOPPY DIAGNOSTIC - BUFFER
2774      BUF      BSS 5000      BUFFER
2775      SPC 2
2776      *****
2777      *
2778      * JOHN 15:8 .....
2779      *
2780      * "'BY THIS IS MY FATHER GLORIFIED, THAT YOU BEAR
2781      * MUCH FRUIT, AND SO PROVE TO BE MY DISCIPLES...'"
2782      *
2783      *****
2784      SPC 5
2785      END RTE

```